

***LES GRANDS CLASSIQUES
DU JEU***

POUR VOTRE

T199/4A

Bob Connors & Sam Edwards



***LES GRANDS CLASSIQUES
DU JEU
POUR VOTRE
TI 99/4A
Bob Connors & Sam Edwards***

*Traduit par
ISABELLE BOCH*



ADDISON-WESLEY EUROPE



InterEditions
87, AVENUE DU MAINE 75014 PARIS

L'édition originale de cet ouvrage a été publiée en Grande Bretagne par
Addison-Wesley Publishers Limited, Londres, sous le titre
Terrific Games for the TI 99/4A. © 1983

Dessin de couverture de Stuart Hughes
Illustrations de Agnes Lehár-Graham

© 1984, **InterÉditions, Paris** et **Addison-Wesley Europe.**

Tous droits réservés. Aucun extrait de ce livre ne peut être reproduit, sous quelque forme ou par quelque procédé que ce soit (machine électronique, mécanique, à photocopier, à enregistrer ou toute autre) sans l'autorisation écrite préalable des Éditeurs.


ISBN 2-7296-0105-8

TABLE DES MATIERES

Le marais des crapauds géants*	11
Monstres intergalactiques	15
Fièvre à Las Vegas*	20
Le perroquet	27
Le Derby de Chantilly*	29
Le cube multicolore*	33
Qui Vive	36
Testez votre mémoire	45
De 1 à 5	48
Escher	52
Les chiffres sans les lettres	57
Dents de la mer	61
Pêle-mêle Shakespearien	65
Explosion*	69
New-York, New York*	75
La clé	79
La boîte noire	83
Chasse au trésor	87
I.T. ou le voyage au centre de la terre*	91
Le loup et les agneaux*	103
Autoroute	107
Formule 1*	109
Tiroir-caisse	115
La traversée*	118
Abordage extra-terrestre*	122
Mini Mancala*	125
Arrêtez-la!	129
B.A. BA = ABBA	131
Voyelles et consonnes	134
Astrologie	136

* Ces jeux sont en BASIC étendu

Ce livre a vu le jour à l'Université de Twente, aux Pays-Bas où nous avons fait la connaissance des joueurs fanatiques suivants: Hermie Hermens, Peter de Vries, Arie Jan Stapel, Hans Rieseboos, Arend Rensink, Mark Sijbenga, Erik Groenhuis, Wim Koolhoven, Ronald Bosch, Berry Hoogstraten, Claudia Dispa, Alex Pelsmaecker et Kasper Boon. Nous les remercions tous pour les longues heures passées avec nous à discuter, créer et écrire les programmes de jeux figurant dans ce livre.




INTRODUCTION

Tout le monde sait que les ordinateurs sont utilisés pour toutes sortes de choses sérieuses - calculs financiers, scientifiques ou traitement de texte, par exemple. Ils peuvent toutefois nous rendre service d'une autre manière. Comme le dit le proverbe, "il faut savoir joindre l'utile à l'agréable", et c'est vrai que nous avons tous besoin de faire fonctionner nos petites cellules grises en dehors des banales préoccupations quotidiennes. C'est pour ça que nous lisons des livres, que nous jouons. La très belle peinture de Bruegel "Jeux d'enfants" illustre la pérennité de ce propos.

Ce livre a pour but de fournir des jeux pour les 7 à 77 ans. On y trouve aussi bien des jeux pour jouer en groupe que des jeux où l'on se confronte seul à la machine. Ces jeux peuvent être des jeux d'action rapide qui vous permettent de tester vos réflexes ou votre capacité à contrôler les nombreux objets mobiles apparaissant sur l'écran, des puzzles et autres casse-têtes qui vous feront réfléchir, des jeux où le joueur se mesure à l'ordinateur et bien sûr quelques jeux de café merveilleux qui associent ces trois tendances. Vous trouverez aussi dans ce livre quelques jeux très intrigants qui n'appartiennent à aucune catégorie. Un certain nombre de jeux se révèlent passionnants bien que leurs listings soient très courts et qu'on puisse donc les entrer en très peu de temps.

En regardant à nouveau "Jeux d'enfants" une pensée vient aussitôt à l'esprit. Si Bruegel vivait aujourd'hui, est-ce que son tableau ressemblerait à ceci ...



LE MARAIS DES CRAPAUDS GEANTS

Attention! Les crapauds géants pullulent dans le marais que vous devez traverser en marchant sur un alignement de radeaux. Au début de la partie, la situation est la suivante:



vous êtes à un coin du marais, et vous devez rejoindre le coin diamétralement opposé.

L'ordinateur vous demande:

BOUGER OU POSER (B OU P)?

Pour mettre un radeau à côté de celui sur lequel vous êtes, tapez P et l'ordinateur vous répond:

DIRECTION?

Tapez N pour Nord, S pour Sud, E pour Est et O pour Ouest et un radeau apparaîtra à l'endroit choisi (le Nord est en haut de l'écran). Pour bouger, taper M et l'ordinateur répond:

DIRECTION?

Comme précédemment, tapez N, S, E ou O, mais attention: n'essayez pas de marcher sur les eaux!

Cela paraît facile mais, de temps en temps, un des crapauds géants jaille du marais pour se précipiter sur vos radeaux. Si vous êtes sur son chemin, il vous gobe. Grâce à Dieu, les crapauds géants ont une mauvaise vue et manquent le plus souvent les radeaux pour retomber dans le marais.

Pour réussir à ce jeu, il vous faut élaborer votre propre stratégie. Plus votre route est longue et compliquée, plus il sera difficile aux crapauds géants de vous attraper mais plus il vous faudra de temps pour traverser (avec le risque qu'un crapaud apparaisse ...).

Tiens, en voilà un qui vient de surgir derrière vous. J'espère qu'il vous manquera!

```
10 REM TI(ETENDU)-LE MARAIS AUX CRAPAUDS
20 CALL CLEAR
30 DIM MA(11,16,2),D$(2)
40 RANDOMIZE
50 CALL CHAR(120,"00667E187E818166")
60 CALL CHAR(130,"18183C5A993C2466")
70 CALL CHAR(140,"FFFFFFFFFFFFFF")
80 CALL CHAR(110,"0000001818000000")
90 MA(10,15,1)=1
100 CALL COLOR(0,4,4):: CALL COLOR(12,7,12)
110 CALL COLOR(13,2,16):: CALL COLOR(14,16,6)
120 CALL COLOR(10,2,4)
130 FOR R=1 TO 10
140 CALL HCHAR(5+R,6,30,15)
150 NEXT R
160 CALL HCHAR(15,20,130)
170 X=10 :: Y=15
180 D$(1)="BOUGER OU POSER?(B OU P)"
190 D$(2)="DIRECTION:?"
210 DISPLAY AT(20,3)BEEP:D$(1)
220 REM ENTREE*****
230 CALL KEY(3,K,S):: IF S=0 THEN 230
240 IF K<>66 AND K<>80 THEN 230
250 ON INT(K/16)-3 GOSUB 300,410
260 TURNS=TURN+1
270 IF RND<.4 AND TURNS>6 THEN GOSUB 480
```

```

280 GOTO 210
290 REM MOUVEMENT*****
300 DISPLAY AT(20,3)BEEP:D$(2)
310 CALL KEY(3,K,S):: IF S=0 THEN 310
320 TX=X-(K=83)+(K=78)
330 TY=Y-(K=69)+(K=79)
340 IF TX=0 AND TY=1 THEN PRINT "VOUS AVEZ REUSSI," :: PRINT "FELICITATIONS" ::
END
350 IF (TX>10)+(TX<1)+(TY>15)+(TY<1)<>0 THEN 390
360 CALL HCHAR(X+5,Y+5,140):: CALL HCHAR(TX+5,TY+5,130)
370 X=TX :: Y=TY
380 IF MA(X,Y,0)=0 THEN PRINT "VOUS VOUS ETES NOYE !" :: END
390 RETURN

```



```

400 REM POSER*****
410 DISPLAY AT(20,3)BEEP:D$(2)
420 CALL KEY(3,K,S):: IF S=0 THEN 420
430 TX=X-(K=83)+(K=78)
440 TY=Y-(K=69)+(K=79)
450 IF (TX>10)+(TX<1)+(TY>15)+(TY<1)=0 THEN CALL HCHAR(TX+5,TY+5,140):: MA(TX,TY,0)=1
460 RETURN
470 REM MOUVEMENT MONSTRE*
480 X1=10 :: Y1=15
490 K=INT(RND*6)

```



```

500 TX1=X1+(K=1)-(K=2)+(K=4):: TY1=Y1+(K=3)-(K=0)+(K=5)
510 IF TX1=X AND TY1=Y THEN PRINT "IL VOUS A GOBE !" :: END
520 IF (TX1>10)+(TX1<1)+(TY1>15)+(TY1<1)<>0 THEN 490
530 IF MA(TX1,TY1,0)=1 THEN CALL HCHAR(X1+5,Y1+5,140):: CALL HCHAR(TX1+5,TY1+5,1
20):: X1=TX1 :: Y1=TY1 :: IF RND<.7 THEN 500 ELSE 490
540 BOL=0
550 FOR C1=-1 TO 1
560 FOR C2=-1 TO 1
570 IF ABS(C1)<>ABS(C2) THEN BOL=BOL+(MA(X1+C1,Y1+C2,1)=1)
580 NEXT C2
590 NEXT C1
600 IF BOL=-3 THEN MA(X1,Y1,1)=1 :: MA(X1,Y1,0)=0
610 IF MA(TX1,TY1,1)=1 THEN 490
620 MA(TX1,TY1,1)=1 :: CALL HCHAR(X1+5,Y1+5,140):: CALL HCHAR(TX1+5,TY1+5,110)
630 RETURN

```

MONSTRES INTERGALACTIQUES

Vous la voyez enfin la Voie Lactée, dans le faisceau du radar x-y-Gz. Dans exactement 2,56 protosecondes vous serez à l'abri dans votre galaxie.

Mais pour l'instant, le danger est là, et en tant que commandant du vaisseau, vous devez agir. Regardez attentivement l'écran XR6. Pas encore d'alerte, la tension monte ... 2,5 protosecondes ... 2 protosecondes ..., 1,5 protoseconde, attention! la voici:

LA CEINTURE D'ASTEROÏDES DE VAN ALLEN

Elles sont nombreuses, ces fameuses ceintures à l'intérieur desquelles sont tapis les redoutables monstres intergalactiques. Aucune arme pour vous défendre contre ces créatures menaçantes, la seule chose à faire est de vous échapper.

Une fois le programme démarré, et la ceinture d'astéroïdes en vue, vous pouvez vous déplacer en appuyant sur:

- ↑ pour aller une case vers le haut
- ← pour aller une case vers la gauche
- pour aller une case vers la droite
- ↓ pour aller une case vers le bas

Appuyez sur la touche jusqu'à ce que le vaisseau bouge (cela peut prendre quelques instants).


Ces terribles monstres intergalactiques approchent inéluctablement. Votre unique chance de survie est de les éviter! La deuxième fois que vous traverserez une ceinture d'astéroïdes de Van Allen, les monstres se montreront plus agressifs et sembleront même prévoir chacun de vos mouvements. S'ils parviennent à vous attaquer cinq fois, c'en est fait de vous et de votre vaisseau.

Bonne chance commandant, dans quelques protosecondes vous serez à bon port ... si Dieu le veut!

```

10 REM TI-MONSTRES INTERGALACTIQUES
20 A1$="0066421818423C00"
30 A2$="007E7E7E7E7E7E00"
40 A3$="18183C5A993C2466"
50 CALL COLOR(1,8,16)
60 CALL COLOR(2,9,16)
70 CALL COLOR(3,2,16)
80 CALL CHAR(35,A1$)
90 CALL CHAR(45,A2$)
100 CALL CHAR(55,A3$)
110 DIM X(12)
120 DIM Y(12)
130 GOTO 1070
140 REM ECRAN*****
150 CALL CLEAR
160 FOR T=1 TO 11
170 CALL HCHAR(5+T,10,45,11)
180 NEXT T
190 CALL SOUND(300,262,0)
200 CALL SOUND(300,330,0)
210 CALL SOUND(300,392,0)
220 CALL SOUND(1000,262,0,330,0,392,0)
230 CALL HCHAR(13,10,35,11)
240 CALL HCHAR(6,15,55)
250 X1=6
260 Y1=15
270 FOR T=1 TO 11
280 X(T)=13
290 Y(T)=9+T
300 NEXT T
310 RETURN
320 REM MOUVEMENT JOUEUR****
330 CALL SOUND(10,880,0)
340 CALL KEY(3,KEY,ST)
350 CALL HCHAR(X1,Y1,45)
360 TURNS=TURNS+1
370 IF KEY=88 THEN 420
380 IF KEY=83 THEN 450
390 IF KEY=68 THEN 480
400 IF KEY=69 THEN 510
410 GOTO 530
420 IF X1=16 THEN 530
430 X1=X1+1
440 GOTO 530

```



```
450 IF Y1=10 THEN 530
460 Y1=Y1-1
470 GOTO 530
480 IF Y1=20 THEN 530
490 Y1=Y1+1
500 GOTO 530
510 IF X1=6 THEN 530
520 X1=X1-1
530 CALL HCHAR(X1,Y1,55)
540 RETURN
550 REM ATTRAPE*****
560 FOR T=1 TO 11
570 IF X1<>X(T) THEN 640
580 IF Y1<>Y(T) THEN 640
590 CALL SOUND(-100,440,0)
600 CALL SOUND(-100,220,0)
610 CALL SOUND(-100,110,0)
```

```

620 CALL SOUND(1000,-4,0)
630 K=K+1
640 NEXT T
650 RETURN
660 REM MOUVEMENT MONSTRES**
670 FOR T=1 TO 11
680 CALL HCHAR(X(T),Y(T),45)
690 IF RND<CHANCE THEN 950
700 A=INT(RND*4)
710 IF A=0 THEN 750
720 IF A=1 THEN 770
730 IF A=2 THEN 790
740 GOTO 810
750 X(T)=X(T)+1
760 GOTO 820
770 X(T)=X(T)-1
780 GOTO 820
790 Y(T)=Y(T)+1
800 GOTO 820
810 Y(T)=Y(T)-1
820 IF X(T)<6 THEN 870
830 IF X(T)>16 THEN 890
840 IF Y(T)<10 THEN 910
850 IF Y(T)>20 THEN 930
860 GOTO 940
870 X(T)=X(T)+1
880 GOTO 1010
890 X(T)=X(T)-1
900 GOTO 1010
910 Y(T)=Y(T)+1
920 GOTO 1010
930 Y(T)=Y(T)-1
940 GOTO 1010
950 X2=X1-X(T)
960 Y2=Y1-Y(T)
970 IF RND<.8 THEN 1000
980 X(T)=X(T)+SGN(X2)
990 GOTO 1010
1000 Y(T)=Y(T)+SGN(Y2)
1010 CALL HCHAR(X(T),Y(T),35)
1020 NEXT T
1030 FOR U=1 TO 11
1040 CALL HCHAR(X(U),Y(U),35)
1050 NEXT U

```

```

1060 RETURN
1070 REM PROGRAMME PRINCIPAL***
1080 CHANCE=CHANCE+.1
1090 GOSUB 140
1100 GOSUB 320
1110 GOSUB 550
1120 GOSUB 660
1130 IF K>4 THEN 1220
1140 IF X1<16 THEN 1100
1150 SCORE=SCORE+POINTS-TURNS-5*K
1160 POINTS=POINTS+20
1170 TURNS=0
1180 GOTO 1080
1190 CALL SOUND(-100,110,0)
1200 CALL SOUND(1000,-4,0)
1210 RETURN
1220 PRINT "FIN DU JEU."
1230 PRINT "VOTRE SCORE : "
1240 PRINT SCORE
1250 PRINT "POUR REJOUER TAPÉZ 'RUN'."
1260 END

```

FIEVRE A LAS VEGAS

Avez-vous déjà vu ces cinglés laisser jusqu'à leur dernier jeton dans des machines rutilantes? Laissez-vous aller à cette douce folie en transformant votre ordinateur en machine à sous tout droit sortie d'un casino de Las Vegas.

Toutes les instructions de jeu apparaîtront à l'écran. A certains moments, vous devrez choisir dans une liste d'options, par exemple:

INSERT, HOLD, PLAY, END

(Pour mémoire, INSERT = introduire la monnaie; HOLD = arrêter la roue; PLAY = tourner la roue; GAMBLE = jouer; COLLECT = ramasser la monnaie; END = fin de partie)

Faites votre choix en tapant la première lettre de l'opération que vous voulez exécuter, par exemple P pour faire jouer la machine. Les roues sont numérotées 1, 2 et 3. Si vous voulez tenir immobile(s) une ou plusieurs roues, tapez H suivi du ou des chiffres correspondants.


Les lumières clignotent et la musique joue pendant que les roues tournent à toute vitesse. Gagné? Non? Aucune importance, la prochaine fois sera la bonne!

```
10 REM TI(ETENDU)-FIEVRE A LAS VEGAS
20 CALL CLEAR :: RANDOMIZE
30 DIM WF(13),R(2,13),JAC(13):: JAC(5)=-1 :: JAC(6)=-1 :: JAC(7)=-1
40 FOR VV=1 TO 13 :: READ WF(VV),R(1,VV),R(2,VV):: NEXT VV
50 DATA 110,0,0,116,0,0,115,0,0,109,0,0,106,8,50,105,8,40,114,7,30
60 DATA 113,5,25,104,3,20,136,3,20,137,2,10,128,2,10,112,0,0
70 DIM JPX(4),JPY(4),JF(4),JCC(4):: FOR JO=1 TO 4 :: READ JPX(JO),JPY(JO),JF(JO)
:: NEXT JO
80 DATA 11,8,110,12,8,116,14,8,115,15,8,109
90 NJ$=CHR$(138)
100 DIM WPX(3),WPY(3),V(4),HVV(3):: WPX(1)=11 :: WPX(2)=13 :: WPX(3)=15 :: FOR I
=1 TO 3 :: WPY(I)=10 :: NEXT I
```

```

110 DIM HPX(3),HPY(3),HF$(3):: FOR I=1 TO 3 :: HPX(I)=WPX(I):: HPY(I)=13 :: NEXT
I
120 H1$=CHR$(107):: NH$=CHR$(108):: HH$=CHR$(96):: H0$=H1$&HH$&H1$&HH$&H1$
130 DIM GPX(4),GPY(4),GV(4):: GPX(1)=22 :: GPY(1)=1
140 GPX(2)=27 :: GPY(2)=2 :: GPX(3)=22 :: GPY(3)=3 :: GPX(4)=17 :: GPY(4)=2
150 NJ=4 :: NH=0 :: NG=0 :: GW=0 :: NI=0 :: NT=-1
160 GOTO 1740
170 REM BLINK S1$/S2$ AND GET IN
180 DISPLAY AT(PLY,PLX):S1$:: CH=0 :: FOR DE=1 TO 10
190 IF CH=0 THEN CALL KEY(0,GET,CH)
200 NEXT DE :: DISPLAY AT(PLY,PLX):S2$:: IF CH=1 THEN RETURN
210 FOR DE=1 TO 10 :: IF CH=0 THEN CALL KEY(0,GET,CH)
220 NEXT DE :: IF CH=1 THEN RETURN
230 GOTO 180
240 REM AJOUT ARGENT
250 FOR AD=SGN(IN)TO IN STEP SGN(IN)
260 DISPLAY AT(1,7):M0+AD;
270 IF SGN(IN)>0 THEN 290
280 CALL SOUND(-50,200,2):: GOTO 300
290 FOR S0=500 TO 700 STEP 100 :: CALL SOUND(-50,S0,2):: NEXT S0
300 NEXT AD :: M0=M0+IN :: RETURN
310 REM ENLEVER DOUBLE
320 D0=0 :: CALL HCHAR(2,3,32,7)
330 FOR J0=1 TO 4 :: CALL HCHAR(JPY(J0),JPX(J0)+2,JF(J0)):: JCC(J0)=0 :: NEXT J0
340 NJ=4 :: RETURN
350 CALL HCHAR(24,3,32,28):: RETURN
360 REM JACKPOT
370 FOR TI=1 TO 4 :: CALL HCHAR(4,3,32,JC)
380 FOR C=1 TO JC :: CALL SOUND(-100,150+20*C,0):: DISPLAY AT(4,C):"J";:: NEXT C
390 NEXT TI :: RETURN
400 REM DEFINITION CARACTERES
410 CALL CHAR(96,"FFFFFFFFFFFFFFFF")
420 CH$="0F0F0F0F0F0F0F0F" :: CALL CHAR(97,CH$)
430 CALL CHAR(60,CH$)
440 CALL COLOR(9,6,1,10,2,16,11,7,16,12,16,6,13,11,16,14,13,16)
450 CALL CHAR(107,"FFBBBBBB3BBBBBB")
460 CALL CHAR(108,"FFFFFFFFFFFFFFFF")
470 A$="1898FF3D3C3CE404"
480 CALL CHAR(109,A$):: CALL CHAR(115,A$)
490 A$="1819FFBC3C3C2720"
500 CALL CHAR(110,A$):: CALL CHAR(116,A$)
510 CALL CHAR(136,"02043C566A566A3C")
520 CALL CHAR(114,"00006C7C7C381000")
530 CALL CHAR(112,"0204387C7C7C3800")

```

```


540 CALL CHAR(104,"02020C3C78783000")
550 CALL CHAR(128,"00406060703C1E00")
560 CALL CHAR(137,"02041C3C7C7C3800")
570 CALL CHARPAT(36,A$):: CALL CHAR(106,A$)
580 CALL CHARPAT(63,A$):: CALL CHAR(120,A$)
590 CALL CHAR(113,"081C3E7F3E1C0800")
600 CALL CHAR(104,"081C2A772A080800")
610 CALL CHAR(105,"0010387C7C103800")
620 CALL CHAR(138,"0000000000000000")
630 RETURN
640 REM **TRACE ECRAN
650 CALL CLEAR :: CALL SCREEN(12)
660 CALL HCHAR(7,12,96,7)
670 FOR I=8 TO 12 :: CALL HCHAR(I,11,96,9):: NEXT I
680 CALL VCHAR(7,20,60)
690 CALL VCHAR(8,20,97,4)
700 CALL HCHAR(12,20,96)
710 CALL HCHAR(13,12,96,7)
720 FOR I=14 TO 19 :: CALL HCHAR(I,13,96,5):: NEXT I
730 CALL HCHAR(20,12,96,7)
740 CALL HCHAR(21,11,96,9)
750 CALL HCHAR(22,10,96,11)
760 DISPLAY AT(1,1):"TOTAL: 0"
770 DISPLAY AT(10,3):CHR$(112);" = 1";
780 FOR VV=12 TO 5 STEP -1
790 DISPLAY AT(23-VV,1):"-";RPT$(CHR$(WF(VV)),2);"-=";R(1,VV);
800 DISPLAY AT(23-VV,20):RPT$(CHR$(WF(VV)),3);"=";R(2,VV);
810 IF JAC(VV)THEN DISPLAY AT(23-VV,27):"+J";
820 NEXT VV :: RETURN
830 REM ** ADAPT VARIABLES
840 NT=NT+1 :: IF WI>0 THEN HB=-1 :: WI=0
850 FOR HO=1 TO 3 :: HF$(HO)=NH$ :: DISPLAY AT(HPY(HO),HPX(HO)):NH$;:: NEXT HO
860 IF DO THEN 910

```

```

870 FOR JO=1 TO 4 :: IF JCC(JO)=0 THEN 900
880 JCC(JO)=JCC(JO)-1 :: IF JCC(JO)>0 THEN 900
890 NJ=NJ+1 :: CALL HCHAR(JPY(JO),JPX(JO)+2,JF(JO))
900 NEXT JO :: GOTO 920
910 DC=DC-1 :: IF DC=0 THEN GOSUB 320
920 IF JA THEN DISPLAY AT(4,JC):" ";; JC=JC-1 :: IF JC=0 THEN JA=0
930 RETURN
940 REM **PAS ASSEZ D'ARGENT
950 HB=-1 :: GOSUB 320
960 JA=0 :: CALL HCHAR(4,3,32,15)
970 GOSUB 350 :: DISPLAY AT(24,1):"INSERT OU END";
980 S1$="INSERT (I)" :: S2$=RPT$(" ",10):: PLX=1 :: PLY=2 :: GOSUB 180
990 IF GET=ASC("P")THEN 980 ELSE RETURN
1000 REM ** HOLD POSSIBLE
1010 DISPLAY AT(24,1):"INSERT, HOLD, PLAY OU END";
1020 S1$=H0$ :: S2$=HF$(1)&HH$&HF$(2)&HH$&HF$(3)
1030 PLX=11 :: PLY=13 :: GOSUB 180
1040 IF GET<49 OR GET>51 THEN RETURN ELSE H0=GET-48
1050 NH=NH+1 :: IF HF$(H0)=NH$ THEN HF$(H0)=H1$ ELSE HF$(H0)=NH$
1060 GOTO 1020

```



```

1070 REM **PAS DE HOLD
1080 GOSUB 350 :: DISPLAY AT(24,1):"INSERT, PLAY OU END";
1090 S1$=RPT$(CHR$(120),3):: S2$=RPT$(CHR$(96),3):: PLX=12 :: PLY=19 :: GOSUB 180
1100 RETURN
1110 REM ** QUE FAIRE DES GAINS?
1120 GOSUB 350 :: IF NOT HB THEN DISPLAY AT(24,1):"HOLD, ";
1130 DISPLAY AT(24,7):"GAMBLE OU COLLECT";
1140 GOTO 1090
1150 REM ** ROUES TOURNANTES
1160 FOR I=7 TO 10 :: CALL HCHAR(I,20,32):: CALL HCHAR(I+1,20,60):: NEXT I
1170 CALL SOUND(-4000,-7,29)

```

```

1180 FOR I=10 TO 7 STEP -1 :: CALL HCHAR(I,20,60):: CALL HCHAR(I+1,20,97):: NEXT
I
1190 FOR WD=1 TO 3 :: IF HF$(WD)=NH$ THEN DISPLAY AT(WPY(WD),WPX(WD)):CHR$(138);
1200 NEXT WD
1210 FOR WD=1 TO 3 :: IF HF$(WD)=H1$ THEN 1240
1220 FI=INT(RND*100+1):: IF FI<5 THEN IF JCC(FI)<>0 THEN V(WD)=13 ELSE V(WD)=FI
:: JCC(FI)=-1 :: GOTO 1240
1230 V(WD)=5-(FI>7)-(FI>10)-(FI>13)-(FI>23)-(FI>36)-(FI>49)-(FI>68)-(FI>87)
1240 FOR DE=2 TO 300 :: NEXT DE :: DISPLAY AT(WPY(WD),WPX(WD)):CHR$(138);
1250 CALL SOUND(-100,300,2):: CALL HCHAR(WPY(WD),WPX(WD)+2,WF(V(WD)))
1260 CALL SOUND(4000,-7,29):: NEXT WD :: CALL SOUND(-1,-2,30):: RETURN
1270 REM **ATTENTION AUX JOCKERS
1280 J=0 :: FOR WD=1 TO 3 :: VV=V(WD)
1290 IF VV>4 THEN FV=VV :: GOTO 1320
1300 JW=WD :: J=J+1 :: IF JCC(VV)>0 THEN 1320
1310 DISPLAY AT(JPY(VV),JPX(VV)):NJ$;:: JCC(VV)=20 :: NJ=NJ-1
1320 NEXT WD :: DO=(NJ=0):: IF DO THEN DC=15 :: DISPLAY AT(2,1):"DOUBLE!";
1330 RETURN
1340 REM ** CALCUL DES GAINS
1350 HV=13 :: ON J+1 GOTO 1360,1380,1360,1410
1360 IF NOT(JA AND FV>10) THEN HV=FV
1370 GOTO 1410
1380 V(0)=V(3):: V(4)=V(1):: IF JA AND V(JW-1)<>V(JW+1) THEN 1410
1390 V(0)=15 :: V(4)=15
1400 IF V(JW+1)>V(JW-1) THEN HV=V(JW-1) ELSE HV=V(JW+1)
1410 FOR WD=1 TO 3 :: IF V(WD)<5 THEN HVV(WD)=HV ELSE HVV(WD)=V(WD)
1420 NEXT WD :: IF HVV(1)<>HVV(2) OR HVV(2)<>HVV(3) THEN 1440
1430 IF JAC(HVV(1)) THEN JA=-1 :: JC=15 :: GOSUB 370
1440 FOR WD=1 TO 3 :: IF HVV(WD)=13 THEN WI=WI+1
1450 NEXT WD :: IF JA THEN WI=10*WI
1460 NS=-(HVV(1)=HVV(2))-(HVV(2)=HVV(3)):: IF NS>0 THEN WI=WI+R(NS,HVV(2))
1470 IF DO THEN WI=2*WI
1480 RETURN
1490 REM ** GAMBLE
1500 DT=1 :: GV(1)=2*WI :: GV(2)=0 :: GV(3)=INT(3*WI/2):: GV(4)=INT(WI/2)
1510 KEY=0 :: NG=NG+1 :: GOSUB 350 :: DISPLAY AT(24,1):"STOP";
1520 FOR LI=1 TO 3 :: CALL HCHAR(LI,16,96,17):: NEXT LI
1530 RR=RR+1 :: IF RR>4 THEN RR=1
1540 DISPLAY AT(GPY(RR),GPX(RR)):STR$(GV(RR));
1550 CALL SOUND(-4000,150+50*RR,2)
1560 IF KEY<>ASC("S") THEN CALL KEY(0,KEY,CH):: GOTO 1580
1570 DT=(1+RND*2)*DT :: FOR DE=1 TO DT :: NEXT DE :: IF DT>150 THEN 1590
1580 CALL HCHAR(GPY(RR),GPX(RR)+2,96,4):: GOTO 1530
1590 FOR LI=1 TO 3 :: CALL HCHAR(LI,16,32,17):: NEXT LI

```

```

1600 GW=GW+GV(RR)-WI :: WI=GV(RR):: CALL SOUND(-1,150+50*RR,2):: RETURN
1610 REM ** FIN DU JEU
1620 CALL CLEAR :: CALL CHARSET :: CALL SCREEN(8)
1630 DISPLAY AT(5,1):"MONTANT :"
1640 IMAGE " F####.##"
1650 DISPLAY AT(7,3):"MIS :";:: DISPLAY AT(7,19):USING 1640:NI
1660 DISPLAY AT(8,3):"RECU :";:: DISPLAY AT(8,19):USING 1640:MO/4
1670 DISPLAY AT(9,3):"MAX EN UNE FOIS:";:: DISPLAY AT(9,19):USING 1640:MM/4
1680 DISPLAY AT(10,3):"GAGNE PAR GAMBLE:";:: DISPLAY AT(10,19):USING 1640:GW/4
1690 DISPLAY AT(12,1):"NOMBRE DE HOLDS: ";NH
1700 DISPLAY AT(13,1):"NOMBRE DE GAMBLES:";NG
1710 DISPLAY AT(14,1):"NOMBRE DE TOURS: ";NT
1720 RETURN

```



```

1730 REM ** PROGRAMME PRINCIPAL**
1740 GOSUB 410 !DEF CAR
1750 GOSUB 650 !ECRAN
1760 GOSUB 840 !ADAPT
1770 ON -2*HB-(MO>1)+1 GOSUB 950,1010,950,1080
1780 DISPLAY AT(5,9):: CA=-(GET=ASC("I"))-2*(GET=ASC("P"))-3*(GET=ASC("E"))
1790 ON CA+1 GOTO 1770,1800,1820,2020
1800 NI=NI+1 :: IN=4 :: GOSUB 250
1810 GOTO 1770
1820 IN=-2 :: GOSUB 250
1830 GOSUB 1160 !ROUES
1840 IF DO THEN 1860
1850 GOSUB 1280 !JOCKERS
1860 GOSUB 1350 !GAINS
1870 IF HF$(1)=H1$ OR HF$(2)=H1$ OR HF$(3)=H1$ THEN 1890
1880 HB=0 :: LW=WI :: IF WI>0 THEN 1910 ELSE 1760
1890 HB=-1 :: IF WI>LW THEN 1910
1900 DISPLAY AT(5,9):" VOUS AVEZ PERDU !" :: CALL SOUND(-600,200,2):: GOTO 1760
1910 DISPLAY AT(5,9):"VOUS AVEZ GAGNE ";:: DISPLAY AT(5,24):WI;:: FOR DU=1 TO WI
:: CALL SOUND(-50,500,1):: CALL SOUND(1,500,30):: NEXT DU

```

```

1920 IF M0<2 THEN HB=-1
1930 GOSUB 1120
1940 CA=-(GET=ASC("H")AND NOT HB)-2*(GET=ASC("G"))-3*(GET=ASC("C"))
1950 ON CA+1 GOTO 1930,2010,1990,1960
1960 IN=WI :: GOSUB 250
1970 IF WI>MM THEN MM=WI
1980 GOTO 1760
1990 HB=-1 :: GOSUB 1500 !GAMBLE
2000 IF WI>0 THEN 1910 ELSE 1900
2010 WI=0 :: GOTO 1760
2020 GOSUB 1620 !END
2030 END

```

LE PERROQUET

Le jeu du Perroquet est un jeu de compétition très simple. Jouez-y un dimanche après-midi pluvieux avec un ami (surtout si celui-ci a pris un repas copieux et bien arrosé!).

Au début du jeu, votre ordinateur préféré vous gratifiera d'un:

ES-TU PRET PERROQUET?

Oui, vous avez bien lu, il vous traite de perroquet! Tapez n'importe quelle touche et l'ordinateur affichera une lettre. Il vous faut alors taper la même lettre, le plus vite possible - d'où le titre du jeu. L'ordinateur vous propose alors une autre lettre et ainsi de suite pendant 10 secondes, à l'issue desquelles votre score sera affiché.

Etes-vous un bon perroquet? Jouez avec vos amis pour savoir qui d'entre-vous mérite le titre de "Super-Perroquet" (mon meilleur score est 19 - mais j'ai pris des cours particuliers au fin fond de la jungle équatoriale!).



```

10 REM TI-PERROQUET
20 CALL CHAR(50,"1C3A3F3C3C3C7C7E")
30 CALL CHAR(51,"7EFEFEFE7C101800")
40 CALL COLOR(3,7,1)
50 P=0
60 RANDOMIZE
70 DATA BOF,BOF,PAS TERRIBLE,PAS TERRIBLE,PERSEVEREZ,PERSEVEREZ,C'EST MIEUX,C'ES
T MIEUX
80 DATA PAS TROP MAL,PAS TROP MAL,BON,BON,TRES BON,TRES BON,BRAVO,BRAVO,EXCELLEN
T,EXCELLENT,VOUS AVEZ ASSEZ JOUE,VOUS ETES LE MEILLEUR
100 CALL CLEAR
110 PRINT "ES-TU PRET PERROQUET?"
120 CALL KEY(3,K,S)
130 IF S=0 THEN 120
140 CALL CLEAR
150 CALL HCHAR(5,15,50)
160 CALL HCHAR(6,15,51)
170 D=INT(RND*26)+97
180 P=P+1
190 IF P=20 THEN 280
200 CALL HCHAR(10,P+1,D)
210 CALL KEY(5,I,S)
220 IF S=0 THEN 250
230 IF I=D THEN 170
240 IF I=D-32 THEN 170
250 B=B+1
260 IF B=150 THEN 280
270 GOTO 210
280 PRINT "FIN"
290 FOR T=1 TO P
300 READ A$
310 NEXT T
320 PRINT A$

```

LE DERBY DE CHANTILLY

Mesdames et Messieurs, ils sont partis! Flèche Rouge a fait un bon départ ... Blond Garçon donne du fil à retordre à son jockey ... tandis que la célèbre Infante est à la corde. L'excitation est à son comble et nous n'en sommes qu'au début! Arrivent au premier virage Mickey Finn en tête, suivi à une encolure de Petit Conzalès ...

Le Derby de Chantilly simule une course de chevaux. Il comporte trois partants qui ont deux tours à parcourir. Au début du jeu, la machine affiche la cote de chaque partant. Puis vous pariez. Tant que vous avez de l'argent vous pouvez continuer à parier - les chevaux, eux, ne sont jamais fatigués.

Après la commande RUN, on lit

DERBY CHANTILLY

et

VOUS AVEZ ... F
SUR QUEL CHEVAL VOULEZ-VOUS PARIER?

VOICI LES COTES

CHEVAL 1 : 5 CONTRE 1
CHEVAL 2 : 4 CONTRE 1
CHEVAL 3 : 3 CONTRE 1

Lorsque vous n'aurez plus d'argent, votre ordinateur vous proposera une méditation sur le thème:

VOUS VIVEZ AU-DESSUS DE VOS MOYENS

Et si vous êtes un peu mieux loti:

VOUS AVEZ ENCORE ... FRANCS

Tout de suite après avoir fait vos paris, vous verrez l'hippodrome avec les chevaux au départ. Dans la partie inférieure de l'écran vos paris et les cotes des chevaux seront affichés.

Et puis vous verrez

START???

Appuyer sur ENTER et ça y est, ils sont partis!


*C'était un petit cheval blanc
Qu'il avait donc du courage!*

```
10 REM TI(ETENDU)-DERBY CHANTILLY
20 RANDOMIZE
30 CALL CLEAR
40 PRINT "DERBY CHANTILLY"
50 PRINT ""
60 PRINT ""
70 CREDIT=100
80 REM PARIS*****
90 SC=0
100 PRINT "VOUS AVEZ ";CREDIT;"F"
110 IF CREDIT=0 THEN 410
120 PRINT "SUR QUEL CHEVAL VOULEZ-VOUS"
130 PRINT "PARIER?"
140 TR=0
150 PRINT "VOICI LES COTES:"
160 FOR A=1 TO 3
170 RA(A)=INT(RND*5+1)
180 PRINT "  CHEVAL";A;" ";RA(A);"CONTRE 1"
190 NEXT A
200 PRINT
210 INPUT PNR
220 IF (PNR<1)+(PNR>3)<>0 THEN 210
230 INPUT "COMBIEN PARIEZ-VOUS ? ":HM
240 IF HM<=CREDIT THEN 280
250 PRINT "VOUS N'AVEZ PAS ASSEZ      D'ARGENT"
```

```

260 GOTO 230
270 REM COURSE*****
280 GOSUB 430
290 GOSUB 670
300 PW=-((PP(1)>PP(2)AND PP(1)>PP(3))+2*(PP(2)>PP(3)AND PP(2)>PP(1))+3*(PP(3)
>PP(1)AND PP(3)>PP(2)))
310 IF TR<4 THEN 290
320 CALL DELSPRITE(ALL)
330 PRINT "CHEVAL VAINQUEUR ";PW
340 IF PW<>PNR THEN 380
350 CREDIT=CREDIT+(RA(PW)-1)*HM
360 PRINT "VOUS AVEZ GAGNE"
370 GOTO 80
380 PRINT "VOUS AVEZ PERDU"
390 CREDIT=CREDIT-HM
400 GOTO 80
410 PRINT "VOUS VIVEZ AU-DESSUS DE VOS MOYENS"
420 END
430 REM INITIALISATION CHEVAUX*****
440 CALL CLEAR
450 CALL CHAR(60,"00040EFF7B5C8200")
460 CALL CHAR(61,"000C1E383E783010")
470 CALL CHAR(88,"AA55AA55AA55AA55")
480 CALL CHAR(62,"01010101010101")
490 MUL=1/RA(1)+1/RA(2)+1/RA(3)
500 MUL=1/MUL
510 CH=RND
520 FOR A=1 TO 3
530 J(A)=0
540 VP(A)=2.7+SGN(RND-0.5)/10
550 IF CH<=0 THEN W(A)=0 :: GOTO 580
560 W(A)=CH*(MUL/RA(A))
570 CH=CH-(MUL/RA(A))
580 CALL SPRITE(#A,60,2,48+24*A,32)
590 CALL SPRITE(#A+6,88,3,48+24*A,120)
600 CALL SPRITE(#A+3,62,2,48+24*A,230)
610 NEXT A
620 REM START
630 FOR A=1 TO 3
640 CALL MOTION(#A,0,VP(A))
650 NEXT A
660 RETURN
670 REM MVT CHEVAUX*****
680 SC=SC+1

```



```

690 FOR A=1 TO 3
700 IF SC>=30 AND W(A)AND PW<>A THEN VP(A)=VP(A)+.5 :: GOTO 720
710 VP(A)=VP(A)+SGN(RND-0.5)/10+(VP(A)>3.0)-(VP(A)<2.5)
720 CALL POSITION(#A,X,PP(A))
730 HV=0
740 IF PP(A)<95 OR PP(A)>=130 THEN 750 ELSE 780
750 IF X>(48+24*A)THEN HV=-1
760 IF X<(48+24*A)THEN HV=1
770 GOTO 800
780 IF PP(A)>112 THEN CALL SPRITE(#A,60,2,X,PP(A),VP(A),VP(A))ELSE CALL SPRITE(#
A,61,2,X,PP(A),-VP(A),VP(A))
790 GOTO 810
800 CALL MOTION(#A,HV,VP(A))
810 F(A)=0
820 IF J(A)THEN 840 ELSE CALL COINC(#A,#A+6,50,J(A))
830 GOTO 860
840 CALL COINC(#A,#A+3,10,F(A))
850 IF F(A)THEN TR=TR+1 :: J(A)=0
860 NEXT A
870 FOR A=1 TO 3
880 CALL POSITION(#A,X,PP(A))
890 NEXT A
900 RETURN

```


LE CUBE MULTICOLORE

Un cube à 2 dimensions? Impossible bien sûr, mais ce jeu, bien que se jouant sur l'écran de l'ordinateur, ressemble beaucoup au célèbre Rubik's Cube.

Voici 25 cases colorées disposées au hasard sur un tableau de 5×5 . Les lignes de ce tableau sont numérotées comme suit:


1					
2					
3					
4					
5					
	6	7	8	9	0

C'est-à-dire que les lignes horizontales sont numérotées de 1 à 5, et les verticales 6, 7, 8, 9 et 0. L'ordinateur vous demande alors d'entrer le numéro d'une ligne et le nombre de fois que vous voulez bouger les cases de cette ligne. Sur une ligne horizontale, les cases se déplacent vers la droite lorsque vous entrez un nombre. Sur une verticale, elles se déplacent vers le bas. Quand une case disparaît au bout d'une rangée, elle réapparaît de l'autre côté comme ceci



Le but de tout cela? Vous devez redisposer les cases de telle sorte que chacune des 5 lignes horizontales soit d'une seule couleur. Simple n'est-ce pas? Mais essayez, pour voir!

```
10 REM TI(ETENDU)-CUBE MULTICOLORE
20 CALL CLEAR
30 RANDOMIZE
40 GOTO 580
50 REM ECRAN*****
60 FOR A1=1 TO 5
70 FOR A2=1 TO 5
80 COLOR(A1,A2)=10*A1+90
90 CALL HCHAR(8+A1,11+A2,COLOR(A1,A2))
100 NEXT A2
110 NEXT A1
120 J$="FFFFFFFFFFFFFFFF"
130 FOR A=1 TO 5 :: CALL CHAR(10*A+90,J$):: NEXT A
140 CALL COLOR(9,16,1)
150 CALL COLOR(10,12,1)
160 CALL COLOR(12,10,1)
170 CALL COLOR(13,13,1)
180 CALL COLOR(14,5,1)
190 FOR T=1 TO 5
200 CALL HCHAR(8+T,11,ASC(STR$(T))): NEXT T
210 DISPLAY AT(14,10):"67890"
220 RETURN
230 REM TIRAGE ALEATOIRE***
240 FOR T=1 TO 25
250 K=INT(RND*10+48)
260 GOSUB 300
270 NEXT T
280 RETURN
290 REM MOUVEMENT*****
300 ON (K-47)GOSUB 480,490,500,510,520,530,540,550,560,570
310 IF ROW<>0 THEN 400
320 FOR A=5 TO 1 STEP -1
330 COLOR(A+1,COLUMN)=COLOR(A,COLUMN)
340 NEXT A
350 COLOR(1,COLUMN)=COLOR(6,COLUMN)
360 FOR A=1 TO 5
370 CALL HCHAR(8+A,11+COLUMN,COLOR(A,COLUMN))
380 NEXT A
```



```


390 RETURN
400 FOR A=5 TO 1 STEP -1
410 COLOR(ROW,A+1)=COLOR(ROW,A)
420 NEXT A
430 COLOR(ROW,1)=COLOR(ROW,6)
440 FOR A=1 TO 5
450 CALL HCHAR(B+ROW,11+A,COLOR(ROW,A))
460 NEXT A
470 RETURN
480 ROW=0 :: COLUMN=5 :: RETURN
490 ROW=1 :: COLUMN=0 :: RETURN
500 ROW=2 :: COLUMN=0 :: RETURN
510 ROW=3 :: COLUMN=0 :: RETURN
520 ROW=4 :: COLUMN=0 :: RETURN
530 ROW=5 :: COLUMN=0 :: RETURN
540 ROW=0 :: COLUMN=1 :: RETURN
550 ROW=0 :: COLUMN=2 :: RETURN
560 ROW=0 :: COLUMN=3 :: RETURN
570 ROW=0 :: COLUMN=4 :: RETURN
580 REM PROGRAMME PRINCIPAL*****
590 GOSUB 60
600 GOSUB 240
610 CALL KEY(5,K,S)
620 IF S=0 OR K<48 OR K>57 THEN 610
630 GOSUB 300
640 GOTO 610
650 END

```


QUI VIVE

Etre sur le qui-vive signifie être vigilant, être sur ses gardes. Ce jeu s'appelle ainsi parce que, pour triompher il faut être à l'affût des situations gagnantes. Les règles sont très simples mais pas le programme qui permet de les appliquer. En fait, c'est une vraie gageure pour un programmeur de jeux.


QUI VIVE a été inventé par Eugène de Wolf et se joue sur un tableau de 5 cases sur 5. Chaque joueur a 5 pions et doit essayer de les disposer pour former une des 7 figures symétriques suivantes




une ligne horizontale




une ligne verticale




une ligne diagonale




une croix droite




une croix oblique



une aile verticale



une aile horizontale



Au début du jeu, le tableau est vide, vous et l'ordinateur y placez à tour de rôle des pions. Les différentes positions sur le tableau sont enregistrées de la façon suivante:

⑤	1	2	3	4	5
④	6	7	8	9	10
③	11	12	13	14	15
②	16	17	18	19	20
①	21	22	23	24	25
	Ⓐ	Ⓑ	Ⓒ	Ⓓ	Ⓔ

Pour placer un pion dans une position déterminée, entrez simplement les coordonnées voulues: par exemple pour la position 21 entrez A1, 12 B3 etc. Vos pions sont indiqués par la lettre P et ceux de l'ordinateur par la lettre C.

Si les 10 pions ont été placés sur le tableau et que personne n'a gagné, le jeu continue, mais en déplaçant les pions pour essayer d'obtenir une position gagnante. Pour déplacer 1 de vos pions, vous devez d'abord

entrer les coordonnées de sa position actuelle, puis celles de la position où vous désirez aller.

Il vous faut conserver vos chances de faire une figure gagnante tout en empêchant tout mouvement victorieux de l'ordinateur.

Ce programme est sûrement un des plus intéressants du présent ouvrage. Il comprend la liste des 42 figures gagnantes, mais aussi plus de 100 situations particulièrement fortes, à partir desquelles un joueur peut réaliser une ou deux figures gagnantes.


Pendant le jeu, l'ordinateur gère cette liste à partir de la situation visible sur le tableau et évalue le meilleur mouvement à faire. Entraînez-vous à ce jeu sur un échiquier ordinaire, avec un de vos amis pour partenaire avant de vous confronter à l'ordinateur! Je vous laisse maintenant découvrir ce jeu passionnant.

```
10 REM TI(ETENDU)-QUI VIVE
20 CALL CLEAR
30 RANDOMIZE
40 DIM SF(41,4),S0(41)
50 DIM D(107,1),DF(15,8)
60 DIM CC(4,1),B(24),C(24)
70 GOTO 1870
80 REM CALCUL DES POINTS DE DF*****
90 GV=0 :: P1=0 :: P2=0 :: C1=0 :: C2=0
100 D1=B(DF(NF,0)):: D2=B(DF(NF,1))
110 IF D1=1 OR D2=1 THEN P1=1
120 IF D1=10 OR D2=10 THEN C1=10
130 D1=B(DF(NF,2)):: D2=B(DF(NF,3))
140 IF D1=1 OR D2=1 THEN P2=1
150 IF D1=10 OR D2=10 THEN C2=10
160 GV=B(DF(NF,4))+B(DF(NF,5))+B(DF(NF,6))
170 P=GV+P1+P2+C1+C2
180 IF GV=30 AND(C1=0 OR C2=0)THEN P=P-10
190 IF GV=3 AND(P1=0 OR P2=0)THEN P=P-1
200 RETURN
210 REM INITIALISATION FIGURES SIMPLES
220 FOR I=0 TO 41 :: READ X$
230 FOR J=1 TO 5
240 SF(I,J-1)=ASC(SEG$(X$,J,1))-65
250 NEXT J :: NEXT I :: RETURN
260 REM AILES(<>)*****
270 DATA UQMSY,PLHNT,KGCIO
280 DATA AGMIE,FLRNJ,KQWSO
```

```

290 DATA AGMQU,BHNRV,CIOSW
300 DATA EIMSY,DHLRX,CGKQW
310 REM CROIX(X)*****
320 DATA ACGKM,BDHLN,CEIMO
330 DATA FLHPR,GIMQS,HJNRT
340 DATA KMQUW,LNRVX,MOSWY
350 REM PLUS(+)*****
360 DATA BFGHL,CGHIM,DHIJN
370 DATA GKLMQ,HLMNR,IMNOS
380 DATA LPQRV,MQRSW,NRSTX
390 REM DIAGONALES(\/)
400 DATA AGMSY,EIMQU
410 REM COLONNES(I)*****
420 DATA AFKPU,BGLQV,CHMRW
430 DATA DINSX,EJOTY
440 REM LIGNES(-)*****
450 DATA ABCDE,FGHIJ,KLMNO
460 DATA PQRST,UVWXY
470 REM INITIALISATION FIGURE DOUBLE (DF)
480 FOR I=0 TO 107 STEP 12 :: READ X#
490 FOR J=0 TO 11 :: FOR K=0 TO 1
500 L=2*J+K+1 :: D(I+J,K)=ASC(SEG$(X#,L,1))-49
510 NEXT K :: NEXT J :: NEXT I
520 RETURN
530 DATA 171:1A1C1E1M101P2>202B2J
540 DATA 393<3=3?3G474:4=4?4A4G4O
550 DATA 4P505B5D5J696<6C6E6M7=7A
560 DATA 7C7I7O7P8>8B8D8J9?9E9K:?
570 DATA :A:E:K:O:P;>;a;D;J<=<C<I
580 DATA =G=I=O>F>H>J?G?K?P0F0J0L
590 DATA AGAIKAMAOAPBHBJBNCICMCP
600 DATA DJDLONEKEMEOFRFWGSGWHTHW
610 DATA IRIXJSJXKTKXLRLYMSMYNTNY

```



```


620 REM CALCUL CHAMPS FIGURE DOUBLE*****
630 TC=4 :: TN=0
640 FOR I=0 TO 4 :: C(SF(D(DF1,0),I))=1 :: NEXT I
650 FOR I=0 TO 4 :: SF1=SF(D(DF1,1),I)
660 IF C(SF1)=1 THEN 680
670 DF(NF,TN)=SF1 :: TN=TN+1 :: GOTO 690
680 DF(NF,TC)=SF1 :: TC=TC+1 :: C(SF1)=0
690 NEXT I
700 FOR I=0 TO 4 :: SF1=SF(D(DF1,0),I)
710 IF C(SF1)=1 THEN DF(NF,TN)=SF1 :: TN=TN+1 :: C(SF1)=0
720 NEXT I :: RETURN
730 REM
740 REM AFFICHAGE TABLEAU*****
750 PRINT " "
760 FOR I=0 TO 4 :: FOR J=0 TO 4
770 F=5*I+J
780 IF B(F)=0 THEN PRINT ".";
790 IF B(F)=1 THEN PRINT "P";
800 IF B(F)=10 THEN PRINT "C";
810 NEXT J :: PRINT 5-I :: NEXT I
820 PRINT " "
830 PRINT "ABCDE"
840 PRINT " "
850 RETURN
860 REM QUI POSSEDE SF*****
870 OM=-1 :: WM=-1
880 FOR SF1=0 TO 41
890 P=B(SF(SF1,0))+B(SF(SF1,1))+B(SF(SF1,2))+B(SF(SF1,3))+B(SF(SF1,4))
900 IF P=0 OR P=10 OR P=20 OR P=30 THEN S0(SF1)=10 :: GOTO 1000
910 IF P<>40 THEN 950
920 GOSUB 1010
930 GOSUB 1050
940 WM=1 :: SF1=41 :: GOTO 1000
950 IF P>20 OR P<3 OR P=11 OR P=12 THEN S0(SF1)=0 :: GOTO 1000
960 S0(SF1)=1
970 IF P=4 AND OM=-1 THEN GOSUB 1010
980 IF P=14 THEN GOSUB 1050
990 IF P=5 THEN SF1=41
1000 NEXT SF1 :: RETURN
1010 REM DETERMINE MOUVEMENT OBLIGATOIRE*****
1020 FOR I=0 TO 4
1030 IF B(SF(SF1,I))=0 THEN OM=SF(SF1,I)
1040 NEXT I :: RETURN
1050 REM DETERMINE PION CLOUE*****

```

```

1060 FOR I=0 TO 4
1070 IF B(SF(SF1,I))<>10 THEN 1110
1080 FOR J=0 TO CC1-1
1090 IF SF(SF1,I)=CC(J,0)THEN CC(J,1)=1
1100 NEXT J
1110 NEXT I
1120 RETURN
1130 REM INTERET DE DF*****
1140 NC=0 :: NP=0 :: NF=0
1150 FOR I=0 TO 15
1160 DF(I,8)=-1
1170 DF(I,7)=0
1180 NEXT I
1190 FOR DF1=0 TO 107
1200 O1=S0(D(DF1,0)):: O2=S0(D(DF1,1))
1210 IF O1<>O2 OR O1=0 OR O2=0 THEN 1330
1220 GOSUB 620
1230 GOSUB 80
1240 IF P=4 OR P=14 THEN DF(NF,7)=1 :: GOTO 1270
1250 IF O1=1 OR P<=DF(NF,8)THEN 1330
1260 DF(NF,7)=10
1270 DF(NF,8)=P
1280 NF=0
1290 FOR I=0 TO 15
1300 IF DF(I,8)>=DF(NF,8)THEN 1320
1310 IF DF(I,7)=0 OR DF(I,7)=10 THEN NF=I
1320 NEXT I
1330 NEXT DF1
1340 RETURN
1350 REM


```



```

1360 REM VALEUR STRATEGIQUE DU MOUVEMENT****
1370 IF WM=1 THEN 1520
1380 MC=0 :: N2=0
1390 FOR NF=0 TO 15
1400 IF DF(NF,7)=0 THEN 1460
1410 GOSUB 80
1420 IF DF(NF,7)=1 THEN 1450
1430 IF P=MC THEN N2=N2+1
1440 IF P>MC THEN MC=P :: N2=1
1450 IF P=4 THEN NF=15
1460 NEXT NF
1470 IF PP=-1 THEN 1520
1480 IF P=4 THEN RETURN
1490 IF MC<MP THEN RETURN
1500 IF MC=MP AND N2<N1 THEN RETURN
1510 IF MC=MP AND N2=N1 AND RND<.5 THEN RETURN
1520 MP=MC :: N1=N2
1530 PP=CP :: PT=CT
1540 RETURN
1550 REM
1560 REM MOUVEMENT JOUEUR*****
1570 IF PC<5 THEN 1610
1580 PRINT "QUEL PION VOULEZ-VOUS BOUGER ?"
1590 GOSUB 1690 :: PT=X
1600 IF B(PT)<>1 THEN PRINT "IMPOSSIBLE" :: GOTO 1580
1610 PRINT "OU VOULEZ-VOUS LE METTRE?"
1620 GOSUB 1690 :: PP=X
1630 IF B(PP)<>0 THEN PRINT "IMPOSSIBLE" :: GOTO 1610
1640 IF PC=5 THEN B(PT)=0
1650 IF PC<5 THEN PC=PC+1
1660 B(PP)=1
1670 GOSUB 740
1680 RETURN
1690 REM ENTREE*****
1700 INPUT X$ :: IF LEN(X$)<>2 THEN 1750
1710 L$=SEG$(X$,1,1):: D$=SEG$(X$,2,1)
1720 IF L$<"A" OR L$>"E" OR D$<"1" OR D$>"5" THEN 1750
1730 X=ASC(L$)-5*VAL(D$)-40
1740 RETURN
1750 PRINT "ENTREE ERRONEE, RECOMMENCEZ" :: GOTO 1700
1760 REM MOUVEMENT ORDINATEUR***
1770 IF CC1<5 THEN PT=CC1 :: CC1=CC1+1 :: GOTO 1830
1780 PRINT "JE PRENDS ";CHR$(65+PT-INT(PT/5)*5);5-INT(PT/5)
1790 B(PT)=0
1800 FOR CI=0 TO 4


```



```

1810 IF CC(CI,0)=PT THEN PT=CI :: CI=4
1820 NEXT CI
1830 PRINT "JE LE METS A ";CHR$(65+PP-INT(PP/5)*5);5-INT(PP/5)
1840 B(PP)=10 :: CC(PT,0)=PP
1850 GOSUB 740
1860 RETURN
1870 REM PROGRAMME PRINCIPAL*****
1880 PRINT "PATIENCE..."
1890 GOSUB 210
1900 GOSUB 470
1910 CC1=0 :: PC=0
1920 PP=12 :: GOTO 2140
1930 FOR I=0 TO 4 :: CC(I,1)=0 :: NEXT I
1940 GOSUB 860
1950 IF P=5 THEN END
1960 IF WM=1 THEN 1960
1970 GOSUB 1130
1980 MP=0 :: N1=0 :: PP=-1
1990 BF=0 :: EF=24
2000 IF OM<>-1 THEN BF=OM :: EF=OM
2010 FOR CP=BF TO EF
2020 IF B(CP)<>0 THEN 2130
2030 IF OM<>-1 THEN 2040
2040 B(CP)=10
2050 IF CC1<5 THEN GOSUB 1360 :: GOTO 2120
2060 FOR CI=0 TO 4
2070 IF CC(CI,1) THEN 2110
2080 CT=CC(CI,0) :: B(CT)=0
2090 GOSUB 1360
2100 B(CT)=10
2110 NEXT CI
2120 B(CP)=0
2130 NEXT CP
2140 GOSUB 1760
2150 IF WM=1 THEN END
2160 GOSUB 1560
2170 PRINT "PATIENCE..."
2180 GOTO 1930

```



TESTEZ VOTRE MEMOIRE

Un des sujets les plus intéressants de la psychologie est, sans conteste, l'étude de la mémoire. Se souvenir de quelque chose, c'est pouvoir retrouver des connaissances acquises antérieurement. Pourquoi, cependant, nous rappelons-nous de certaines choses et en oublions-nous d'autres? On peut reconnaître quelqu'un qu'on n'a pas vu depuis des années, ou se remémorer un air dont on n'a entendu que quelques notes, tout en demeurant incapable de se souvenir d'un numéro de téléphone. En réalité, il y a 2 sortes de mémoire: la mémoire à court terme (MCT) et la mémoire à long terme (MLT). Si vous désirez en savoir plus sur ce sujet, nous vous invitons à lire l'"Introduction à la psychologie" de Hilgard et Atkinson.

Ce jeu permet de tester votre mémoire à court terme. Jouez-y entre amis et en famille pour savoir qui possède la meilleure mémoire. vous verrez apparaître sur l'écran ce menu:

VOULEZ-VOUS JOUER AVEC


1. DES LETTRES?
2. DES CHIFFRES?
3. 0 ET 1?
4. OU STOP?

Maintenant faites votre choix en entrant 1, 2, 3 ou 4, et choisissez un niveau de difficulté. Supposons que vous entriez 1 pour jouer avec des lettres. Une lettre va apparaître sur l'écran, mais très brièvement. Tapez-la. L'ordinateur vous montre ensuite 2 lettres que vous devez taper à votre tour, puis trois et ainsi de suite. Evidemment, plus le nombre de lettres augmente, plus il devient difficile de se les rappeler toutes. Quelle est la plus longue rangée de caractères dont vous pouvez vous souvenir? Pouvez-vous battre notre record, qui est de 8?


```

10 REM TI-TESTEZ VOTRE MEMOIRE
20 CALL CLEAR
30 RANDOMIZE
40 B=0
50 R$=""
60 PRINT "VOULEZ-VOUS JOUER AVEC:"
70 PRINT "1:DES LETTRES?"
80 PRINT "2:DES CHIFFRES?"
90 PRINT "3:0 ET 1?"
100 PRINT "4:OU STOP?"
110 CALL KEY(5,K,S)
120 IF S=0 THEN 110
130 IF K>57 THEN 110
140 IF K<48 THEN 110
150 K=K-48
160 ON K GOSUB 420,450,480,510
170 INPUT "NIVEAU ?(1-6)":DIF
180 REM AFFICHAGE*****
190 B=B+1
200 R$=""
210 CALL CLEAR
220 FOR CC1=1 TO B
230 A=INT(RND*NR)+W
240 CALL HCHAR(12,16,A)
250 R$=R$&CHR$(A)
260 FOR CC2=1 TO 100/DIF
270 NEXT CC2

```



```

280 CALL HCHAR(12,16,31)
290 FOR CC2=1 TO 100/DIF
300 NEXT CC2
310 NEXT CC1
320 REM ENTREE REPONSE*****
330 INPUT "REPONSE: ":AN$
340 IF AN$=R$ THEN 190
350 PRINT "FAUX"
360 PRINT "VOTRE SCORE:";B-1
370 PRINT "LA REPONSE ETAIT:";R$
380 PRINT "APPUYEZ SUR N'IMPORTE QUELLETOUCHE"
390 CALL KEY(5,K,S)
400 IF S=0 THEN 390
410 GOTO 20
420 NR=26
430 W=65
440 RETURN
450 NR=10
460 W=48
470 RETURN
480 NR=2
490 W=48
500 RETURN
510 PRINT "FIN DU JEU"
520 END

```

DE UN A CINQ

Ce jeu irritant, que l'on peut trouver sur une calculatrice CASIO, fait appel à nos capacités de logique. Il se joue sur un tableau de 9 positions numérotées comme suit

1	2	3
4	5	6
7	8	9

Si, à partir d'une position, on trace une ligne verticale et une ligne horizontale, on barre cinq positions. Par exemple des lignes tirées à partir de la position 4, passeront aussi sur les positions 1, 5, 6 et 7.

1	2	3
4	5	6
7	8	9

Les positions sur le tableau sont occupées par des chiffres tirés au hasard entre 0 et 5. Si vous désignez une des positions (en tapant son numéro), le chiffre placé sur cette position ainsi que tous les chiffres sur

les lignes horizontales et verticales traversant cette position, sont augmentées de un (sauf 5 qui devient 0).

Par exemple, si nous avons:

2	2	4
1	5	2
1	5	4

et que nous entrons 4, le tableau va se transformer comme suit:

3	2	4
2	0	3
2	5	4

Le jeu est fini lorsque le tableau a pour valeur:

0	0	0
0	0	0
0	0	0

```

10 REM TI-DE UN A CINQ
20 RANDOMIZE
30 REM AFFICHAGE*****
40 CALL CLEAR
50 IL$="234WERASD"
60 FOR T3=0 TO 2
70 FOR T4=0 TO 2
80 CALL HCHAR(T3+10,T4+13,ASC(STR$(A(T3,T4))))
90 CALL HCHAR(T3+10,T4+19,ASC(SEG$(IL$,3*T3+T4+1,1)))
100 NEXT T4
110 NEXT T3
120 FOR T1=0 TO 10
130 I=INT(RND*9)+1
140 GOSUB 250
150 NEXT T1
160 M=0
170 REM ENTREE*****
180 CALL KEY(1,I,STA)
190 IF (I<1)+(I>9)<>0 THEN 180
200 IF STA=0 THEN 180
210 M=M+1
220 GOSUB 250
230 GOTO 180
240 REM EXECUTION MOUVEMENT***
250 X=INT((9-I)/3+1)
260 Y=((I-1)/3-INT((I-1)/3))*3+1
270 FOR T2=0 TO 2
280 A(X-1,T2)=A(X-1,T2)+1
290 A(T2,Y-1)=A(T2,Y-1)+1
300 NEXT T2
310 A(X-1,Y-1)=A(X-1,Y-1)-1
320 T=0
330 FOR T3=0 TO 2
340 FOR T4=0 TO 2
350 IF A(T3,T4)<>6 THEN 370
360 A(T3,T4)=0
370 IF A(T3,T4)<>0 THEN 390
380 T=T+1
390 CALL HCHAR(T3+10,T4+13,ASC(STR$(A(T3,T4))))
400 NEXT T4
410 NEXT T3
420 IF T<9 THEN 430 ELSE 450
430 RETURN

```


```
440 REM PROBLEME RESOLU****  
450 PRINT "TERRIBLE !"  
460 PRINT "REUSSI EN ";" COUPS"  
470 END
```

ESCHER


L'oeuvre du graveur hollandais M.C. Escher (1902-1972) basée sur des formes symétriques et mathématiques, est très en vogue chez les programmeurs.

Avec l'aide de l'ordinateur, vous pouvez produire vos propres dessins en utilisant les mêmes principes qu'Escher.


Observez le graphisme suivant:




Il est construit suivant le modèle d'une mosaïque du palais de l'Alhambra en Espagne. Si vous le regardez de près vous vous apercevrez qu'il y a à la base un motif unique et qu'en interchangeant les couleurs noir et blanc, on obtient le même dessin après rotation de 90 degrés. Escher a étudié ces graphismes et inventé un jeu basé sur eux. Il a imaginé un motif dessiné comme ceci



qu'il a fait tourner 4 fois, de 90 degrés



et arrangé pour former un dessin continu.



Ce programme agit de façon très similaire. Après que vous ayez rempli une case "de base" avec des symboles graphiques, l'ordinateur effectue une rotation et la déplace pour remplir l'écran avec votre propre "Escher".

Au démarrage, l'ordinateur affiche les formes qui peuvent être utilisées pour composer votre case de base. Puis il demande:

TAILLE DE LA CASE DE BASE (3-8)

Si vous entrez 3, par exemple, vous aurez neuf positions à remplir. Quand ce remplissage est terminé, l'ordinateur affiche:

DIRECTION DU CARRE (0-3)

Chaque nombre correspond à une des directions de base. Cette question sera répétée à plusieurs reprises, après quoi l'ordinateur affichera votre création. Etonnant, non?

```

10 REM TI-ESCHER
20 DATA 0000000000000000,0000000000000000,0000000000000000,0000000000000000
30 DATA 8080808080808080,FF00000000000000,0101010101010101,00000000000000FF
40 DATA 2020202020202020,0000FF0000000000,0404040404040404,0000000000FF0000
50 DATA 1010101010101010,000000FF00000000,8080808080808080,00000000FF000000
60 DATA C0C0C0C0C0C0C0C0,FFFF000000000000,0303030303030303,000000000000FFFF
70 DATA F0F0F0F0F0F0F0F0,FFFFFFFF00000000,0F0F0F0F0F0F0F0F,00000000FFFFFFF
80 DATA 8040201008040201,0102040810204080,8040201008040201,0102040810204080
90 DATA 101020C000000000,8080403000000000,0000000030408080,00000000C0201010
100 DATA FFFFCF8F0E0C080,FF7F3F1F0F070301,0103070F1F3F7FFF,80C0E0F08FCFEFF
110 DATA F0F0F0F000000000,0F0F0F0F00000000,00000000F0F0F0F0,00000000F0F0F0F0
120 DATA 00183C7E7E3C1800,00183C7E7E3C1800,00183C7E7E3C1800,00183C7E7E3C1800
130 DATA 8142241818244281,8142241818244281,8142241818244281,8142241818244281
140 DATA 08080808FF080808,10101010FF101010,101010FF10101010,080808FF08080808
150 DATA AA55AA55AA55AA55,55AA55AA55AA55AA,AA55AA55AA55AA55,55AA55AA55AA55AA
160 DATA F0F0F0F0F0F0F0F0,0F0F0F0FF0F0F0F0,F0F0F0F0F0F0F0F0,0F0F0F0FF0F0F0F0
170 FOR TEL=96 TO 155
180 READ X$
190 CALL CHAR(TEL,X$)
200 NEXT TEL
210 REM ENTREE CASE DE BASE
220 PRINT "VOUS DISPOSEZ DES CARACTERESSUIVANTS"
230 FOR TEL=96 TO 155 STEP 4
240 PRINT TEL;
250 CALL HCHAR(24,10,TEL)
260 PRINT
270 NEXT TEL
280 INPUT "TAILLE DE LA CASE DE BASE (3-8)":GR
290 IF (GR<3)+(GR>8)<>0 THEN 280
300 HH=INT(24/GR)
310 HHH=HH*HH
320 DIM D(64)
330 DIM RI(64)

```

```

340 I1=0
350 FOR CC=1 TO GR*GR
360 PRINT "NO GRAPHIQUE :";CC;
370 INPUT D(CC)
380 IF (D(CC)<96)+(D(CC)>155)+(D(CC)/4-INT(D(CC))/4<>0)<>0 THEN 360
390 NEXT CC
400 REM ENTREE DIRECTION***
410 FOR CC=1 TO HHH
420 PRINT "DIRECTION DU CARRE ";CC;"(0-3): ";
430 INPUT RI(CC)
440 IF (RI(CC)<0)+(RI(CC)>3)<>0 THEN 420
450 NEXT CC
460 REM TRACE DESSIN ESCHER*
470 CALL CLEAR
480 AAA=0
490 FOR OB1=1 TO HH*GR STEP GR
500 FOR OB2=4 TO HH*GR STEP GR
510 AAA=AAA+1
520 I=0
530 ON RI(AAA)+1 GOSUB 580,640,700,760
540 NEXT OB2
550 NEXT OB1
560 GOTO 560
570 END
580 FOR C1=OB1 TO OB1+GR-1
590 FOR C2=OB2 TO OB2+GR-1
600 GOSUB 820
610 NEXT C2
620 NEXT C1
630 RETURN
640 FOR C2=OB2+GR-1 TO OB2 STEP -1
650 FOR C1=OB1 TO OB1+GR-1
660 GOSUB 820
670 NEXT C1
680 NEXT C2
690 RETURN
700 FOR C1=OB1+GR-1 TO OB1 STEP -1
710 FOR C2=OB2+GR-1 TO OB2 STEP -1
720 GOSUB 820
730 NEXT C2
740 NEXT C1
750 RETURN
760 FOR C2=OB2 TO OB2+GR-1
770 FOR C1=OB1+GR-1 TO OB1 STEP -1

```

```
780 GOSUB 820
790 NEXT C1
800 NEXT C2
810 RETURN
820 I=I+1
830 CALL HCHAR(C1,C2,D(I)+RI(AAA))
840 RETURN
```

LES CHIFFRES SANS LES LETTRES

Voici un jeu simple qui vous permettra de tester vos capacités en calcul mental. L'ordinateur affiche une liste de 7 nombres et un 8e nombre à atteindre. Vous devez choisir 2 de ces 7 nombres et une des opérations suivantes:

addition +
soustraction —
multiplication *
division /

Pour ce faire, vous tapez l'opération telle que vous voulez qu'elle soit exécutée. Par exemple si vous entrez $7 + 18$ l'ordinateur calcule $7 + 18 = 25$. Les nombres 7 et 18 sont ensuite remplacés dans la liste initiale par 25. Choisissez à nouveau deux nombres et une opération. Le but est de terminer avec un nombre dans votre liste égal ou au moins très proche du nombre à atteindre, avant que le temps imparti soit écoulé.

Quand votre temps se sera écoulé, l'ordinateur additionnera les nombres de la liste et affichera la différence entre ce total et le nombre à atteindre. Vous devez faire en sorte que cette différence soit la plus petite possible.

Tout le monde peut participer à ce jeu, puisqu'il est très simple. Vous aurez peut être l'occasion de découvrir qu'il y a un génie du calcul mental parmi vos proches.

```
10 REM TI(ETENDU)-CHIFFRES SANS LETTRES
20 RANDOMIZE
30 FOR B=1 TO 4
40 NR(B)=INT(RND*9+1)
50 NEXT B
60 NR(5)=INT(RND*5+1)*10
70 NR(6)=INT(RND*4+1)*25
80 TARGET=INT(RND*600+200)
90 TI=25
100 GOTO 930
```

```

110 REM*ECRAN
120 CALL CLEAR
130 PRINT "ESSAYEZ D'ATTEINDRE CE      NOMBRE: ";TARGET
140 FOR B=1 TO 6
150 PRINT NR(B);
160 NEXT B
170 PRINT
180 RETURN
190 REM*ENTREE
200 PRINT "OPERATION: ";
210 O$=""
220 CALL HCHAR(12,31,TI+97)
230 CALL KEY(3,K,S)
240 D=D+1
250 IF D<50 THEN 310
260 TI=TI-1
270 IF TI<0 THEN 460
280 CALL SOUND(10,880,0)
290 D=0
300 CALL HCHAR(12,31,TI+97)
310 IF S=0 THEN 230
320 IF K=13 THEN 460
330 IF K=8 THEN 390
340 IF K<42 THEN 220
350 IF K>57 THEN 220
360 IF K=44 THEN 220
370 C$=CHR$(K)
380 PRINT C$;
390 IF K<>8 THEN 440
400 IF LEN(O$)=0 THEN 220
410 PRINT "/";
420 O$=SEG$(O$,1,LEN(O$)-1)
430 GOTO 220
440 O$=O$&C$
450 GOTO 220
460 CALL HCHAR(12,31,31)
470 PRINT
480 RETURN
490 REM*CALCUL
500 C$="+ "
510 GOSUB 690
520 N=N1+N2
530 IF P<>0 THEN 680
540 C$="- "

```



```

550 GOSUB 690
560 N=N1-N2
570 IF P<>0 THEN 680
580 C$="*"
590 GOSUB 690
600 N=N1*N2
610 IF P<>0 THEN 680
620 C$="/"
630 GOSUB 690
640 IF P=0 THEN 680
650 N=N1/N2
660 RETURN
670 ERR=1
680 RETURN
690 REM DETERMINER NOMBRES
700 P=POS(0$,C$,1)
710 IF P=0 THEN 740

```

```

720 N1=VAL(SEG$(0$,1,P-1))
730 N2=VAL(SEG$(0$,P+1,LEN(0$)-P))
740 RETURN
750 REM*ENTREE*
760 T1=0
770 T2=0
780 FOR B=1 TO 6
790 IF N1<>NR(B) THEN 830
800 IF T1<>0 THEN 830
810 T1=B
820 GOTO 850
830 IF N2<>NR(B) THEN 850
840 T2=B
850 NEXT B
860 IF T1=0 THEN 880
870 IF T2<>0 THEN 900
880 ERR=1
890 RETURN
900 NR(T1)=N
910 NR(T2)=0
920 RETURN
930 REM**PROGRAMME PRINCIPAL**
940 GOSUB 110
950 GOSUB 190
960 IF T1<0 THEN 1110
970 ERR=0
980 GOSUB 490
990 IF ERR=1 THEN 1140
1000 GOSUB 750
1010 IF ERR=1 THEN 1140
1020 DIFF=1000
1030 FOR B=1 TO 6
1040 RS=ABS(TARGET-NR(B))
1050 IF RS>DIFF THEN 1070
1060 DIFF=RS
1070 NEXT B
1080 IF DIFF>0 THEN 940
1090 PRINT "BRAVO! VOUS AVEZ REUSSI!"
1100 END
1110 PRINT "VOUS AVEZ ECHOUE!"
1120 PRINT "LA DIFFERENCE EST";DIFF
1130 END
1140 PRINT "ENTREE ERRONEE!"
1150 GOTO 950

```

LES DENTS DE LA MER

Cela fait 5 jours que vous naviguez en fouillant les 7 mers. Soudain à la surface miroitante de l'océan, un léger clapotis - voilà votre gibier, cette terreur des profondeurs - le requin.

Pour déplacer votre bateau utilisez les touches:

Z	E	R
S		D
W	X	C

Par exemple taper E déplace votre bateau vers le nord, c'est-à-dire vers le haut de l'écran. Si vous vous écartez trop du requin, on vous dira

DESOLE; VOUS ETES ALLE TROP LOIN


Quand vous pensez être assez près, tirez sur le requin en tapant la touche d'espacement. Si vous avez atteint votre cible vous verrez le message suivant

TRES BIEN! FELICITATIONS!

Malheureusement vous n'avez que 200 ergs à utiliser. Déplacer votre bateau et faire feu vous coûte de l'énergie. Si vous usez toute votre énergie, l'ordinateur vous déclare

VOUS N'AVEZ PLUS D'ENERGIE

A la fin du jeu, vous saurez ce que vous valez comme chasseur de requins.



```


10 REM TI-LES DENTS DE LA MER
20 DIM C$(3)
30 NG=0
40 TR=0
50 MR=0
60 FOR COM=1 TO 3
70 READ C$(COM)
80 NEXT COM
90 DATA DESOLE; VOUS ETES ALLE TROP LOIN
100 DATA TRES BIEN ! FELICITATIONS !
110 DATA VOUS N'AVEZ PLUS D'ENERGIE
120 GOTO 830
130 REM INITIALISATION****
140 CALL CLEAR
150 PRINT "          ZER"
160 PRINT "          S D"
170 PRINT "          WXC"
180 FOR I=1 TO 8
190 PRINT
200 NEXT I
210 NG=NG+1
220 SX=INT(RND*70-35)
230 SY=INT(RND*70-35)
240 DI=ABS(SX)+ABS(SY)
250 SD=DI
260 MD=DI
270 DATA 200,0,0,0,0
280 READ EN,HI,NS,VX,VY
290 RETURN
300 REM ENTREE VITESSE****
310 CALL KEY(3,K,S)

```

```

320 VX=3*((K=90)+(K=83)+(K=87)-(K=82)-(K=68)-(K=67))
330 VY=3*((K=87)+(K=88)+(K=67)-(K=90)-(K=69)-(K=82))
340 SH=(K=32)
350 RETURN
360 REM CALCUL*****
370 SX=SX-VX
380 SY=SY-VY
390 SX=SX+SGN(SX)
400 SY=SY+SGN(SY)
410 EN=EN+(VX<>0)+(VY<>0)-1
420 IF SH=0 THEN 460
430 NS=NS+1
440 HI=(RND*DI<2)
450 EN=EN-2*DI
460 EN=-(EN>0)*EN
470 IF HI THEN 500
480 DI=ABS(SX)+ABS(SY)
490 MD=-(MD<DI)*MD-(MD>=DI)*DI
500 RETURN
510 REM SORTIE*****
520 IF SH=0 THEN 550
530 CALL SOUND(100,110,0)
540 GOTO 560
550 CALL SOUND(100,3000-25*DI,0)
560 X$=STR$(EN)
570 D=LEN(X$)
580 FOR T=1 TO 4
590 IF T<=D THEN 600 ELSE 620
600 CALL HCHAR(2,14+T,ASC(SEG$(X$,T,1)))

```



```

610 GOTO 630
620 CALL HCHAR(2,14+T,32)
630 NEXT T
640 RETURN
650 REM FIN DU JEU*****
660 PRINT C0$(CA)
670 PRINT "DISTANCE DE DEPART";SD;"M"
680 PRINT "NOMBRE DE TIRS:";NS
690 PRINT "VOUS L'AVEZ APPROCHE A";MD;"M"
700 RE=50-DI
710 IF HI=0 THEN 740
720 PRINT "VOUS L'AVEZ EU DE:";DI;"M"
730 RE=50+EN/4+SD/10
740 TR=TR+RE
750 IF RE<MR THEN 770
760 MR=RE
770 PRINT "GAIN (DE 0 A 100):"
780 PRINT "CE JEU:";INT(RE)
790 PRINT "MOYENNE SUR";NG;"JEUX:";INT(TR/NG)
800 PRINT "MAXIMUM:";INT(MR)
810 RETURN
820 REM PROGRAMME PRINCIPAL*****
830 GOSUB 130
840 GOSUB 300
850 GOSUB 360
860 GOSUB 510
870 CA=-3*(EN<=0)-2*HI-(DI>100)
880 CA=CA+3*(CA>3)
890 IF CA=0 THEN 840
900 GOSUB 650
910 INPUT "ON REJOU?(O/N)":AN$
920 IF SEG$(AN$,1,1)="O" THEN 940
930 END
940 RESTORE 270
950 GOTO 830

```

PÊLE-MÊLE SHAKESPEARIEN

Ce jeu associe Shakespeare, un des plus grands auteurs du monde, et les échecs, le jeu le plus prestigieux, pour former un pêle-mêle étonnant. Des lettres sont disposées sur un échiquier de la façon suivante.

T	O		B	E		O	R
N	O	T		T	O		B
E		T	H	A	T		I
S		T	H	E		Q	U
E	S	T	I	O	N	.	
W	I	L	L	I	A	M	
S	H	A	K	E	S	P	E
A	R	E		I	6	O	3


Une pièce, le cavalier, se déplace sur l'échiquier, comme lors d'une partie d'échecs habituelle. En sautant d'une case à l'autre, les lettres ou symboles dessinés sur les cases sont permutés bouleversant ainsi le texte de l'échiquier.

L'ordinateur vous demande:

NIVEAU:

Entrez le nombre de mouvements que vous voulez voir faire par le cavalier. Naturellement plus ce nombre est élevé, plus l'échiquier s'en trouve chamboulé.

L'échiquier vous est montré après les déplacements du cavalier. Votre tâche est de débrouiller le texte en redéplaçant le cavalier sur l'échiquier. Entrez un nombre de 1 à 8 pour bouger le cavalier, comme indiqué ci-après.

	1		2	
8				3
				
7				4
	6		5	

```


10 REM TI-PELE MELE SHAKESPEARIE
20 DIM HX(8),HY(8),B$(64),T$(8)
30 RANDOMIZE
40 GOTO 560
50 REM INITIALISATION*****
60 X0=1
70 Y0=1
80 FOR K=1 TO 8
90 READ T$(K)
100 FOR J=1 TO 8
110 B$((K-1)*8+J)=SEG$(T$(K),J,1)
120 NEXT J
130 NEXT K
140 FOR K=1 TO 8
150 READ HX(K),HY(K)
160 NEXT K
170 RETURN
180 REM MOUVEMENT CAVALIER*
190 DONE=0
200 IF (XS+X0)>8 THEN 350
210 IF XS+X0<1 THEN 350
220 IF YS+Y0>8 THEN 350
230 IF YS+Y0<1 THEN 350
240 DONE=1
250 XN=X0+XS
260 YN=Y0+YS
270 REM CHANGEMENT LETTRES*****
280 P=XN+(YN-1)*8
290 V=X0+(Y0-1)*8
300 W$=B$(P)
310 B$(P)=B$(V)

```

```

320 B$(V)=W$
330 XO=XN
340 YO=YN
350 RETURN
360 REM AFFICHAGE*****
370 CALL CLEAR
380 PRINT
390 PRINT "12345678"
400 FOR K=1 TO 8
410 FOR J=1 TO 8
420 PRINT B$((K-1)*8+J);
430 NEXT J
440 PRINT K
450 NEXT K
460 PRINT
470 PRINT " 1 2 "
480 PRINT "8  3"
490 PRINT "   "
500 PRINT "7  4"
510 PRINT " 6 5 "
520 PRINT
530 PRINT
540 RETURN
550 REM PROGRAMME PRINCIPAL****
560 GOSUB 50
570 INPUT "NIVEAU=":L
580 FOR K=1 TO L

```



```


590 M=INT(RND*8+1)
600 XS=HX(M)
610 YS=HY(M)
620 GOSUB 180
630 IF DONE=0 THEN 590
640 NEXT K
650 GOSUB 360
660 PRINT "ANCIEN X=";X0
670 PRINT "ANCIEN Y=";Y0
680 GOSUB 760
690 GOSUB 180
700 IF DONE=0 THEN 680
710 GOTO 650
720 DATA "TO BE OR","NOT TO B",E THAT I","S THE QU"
730 DATA "ESTION. ","WILLIAM ","SHAKESPE","ARE 1603"
740 DATA 2,-1,2,1,-1,2,1,2,-2,1,-2,-1,-1,-2,1,-2
750 END
760 REM ENTREE*****
770 INPUT Q
780 IF (Q<1)+(Q>8)<>0 THEN 770
790 XS=(Q=1)+(Q=6)+2*(Q=8)+2*(Q=7)-(Q=2)-(Q=5)-2*(Q=3)-2*(Q=4)
800 YS=(Q=8)+(Q=3)+2*(Q=1)+2*(Q=2)-(Q=7)-(Q=4)-2*(Q=6)-2*(Q=5)
810 RETURN

```


EXPLOSION

La plupart des versions de ce jeu sont conçues pour 2 ou plusieurs participants. Avec ce programme, vous allez pouvoir vous mesurer avec l'ordinateur lui-même. Y êtes-vous prêt?


Explosion se joue sur une grille de 3×3 ou 4×4 cases. L'ordinateur vous demande tout d'abord d'entrer la taille de la grille que vous voulez. Chaque case peut contenir un nombre maximum de pions, égal au nombre de cases qui lui sont directement adjacentes. Ceci signifie que les coins de la grille peuvent contenir 2 pions,



les cases se trouvant sur le bord 3 pions



et celles du centre 4 pions



On vous attribue à vous et à l'ordinateur des pions de couleur différente que vous placez à tour de rôle sur la grille. Vous pouvez mettre un pion sur n'importe quelle case vide, ou sur une case qui contient déjà un ou plusieurs de vos pions. Une case explose lorsqu'elle atteint sa capacité limite de pions. Elle se vide alors de ses pions qui se répartissent sur chaque case voisine, s'emparant des pions de l'adversaire qui s'y trouvent.

Plus le jeu progresse, plus les explosions sont importantes. A la fin toute la grille explose en une seule couleur. Si c'est la vôtre, vous avez gagné.

```
10 REM TI(ETENDU)-EXPLOSION
20 RANDOMIZE
30 DIM RB(5,5),SB(5,5),ST(5,5)
40 GOTO 1620
50 FOR X=1 TO SI
60 FOR Y=1 TO SI
70 SB(X,Y)=RB(X,Y)
80 NEXT Y
90 NEXT X
100 RETURN
110 FOR X=1 TO SI
120 FOR Y=1 TO SI
130 RB(X,Y)=SB(X,Y)
140 NEXT Y
150 NEXT X
160 RETURN
```

```

170 REM **INITIALISATION**
180 INPUT "TAILLE DE LA GRILLE (3 OU 4)":SI
190 IF SI<3 THEN 180
200 IF SI>4 THEN 180
210 FOR X=1 TO SI
220 FOR Y=1 TO SI
230 ST(X,Y)=4+(X=1)+(X=SI)+(Y=1)+(Y=SI)
240 NEXT Y
250 NEXT X
260 INPUT "QUI DEMARRE,VOUS(V) OU MOI(M)?":AN$
270 CM=(SEG$(AN$,1,1)="V")
280 CALL CLEAR
290 FOR I=1 TO SI
300 CALL HCHAR(3,3*I+11,I+48)
310 CALL HCHAR(3,3*I+10,88)
320 NEXT I
330 FOR Y=1 TO SI
340 CALL HCHAR(2*Y+4,8,89)
350 CALL HCHAR(2*Y+4,9,Y+48)
360 FOR X=1 TO SI
370 A=X
380 B=Y
390 GOSUB 1710
400 NEXT X
410 NEXT Y
420 DISPLAY AT(17,1):"X="
430 DISPLAY AT(18,1):"Y="
440 RETURN
450 REM **ENTREE ET EXECUTION MOUVEMENT**
460 CALL HCHAR(17,5,32)
470 CALL HCHAR(18,5,32)
480 CALL SOUND(50,1000,0)
490 CALL KEY(3,MX,S)
500 IF S=0 THEN 490
510 CALL HCHAR(17,5,MX)
520 MX=MX-48
530 IF MX<1 OR MX>SI THEN 490
540 FOR I=0 TO 40
550 NEXT I
560 CALL KEY(3,MY,S)
570 IF S=0 THEN 560
580 CALL HCHAR(18,5,MY)
590 MY=MY-48
600 IF MY<1 OR MY>SI THEN 490

```

```


610 IF RB(MX,MY)<0 THEN 490
620 GOSUB 50
630 X=MX
640 Y=MY
650 DI=-1
660 GOSUB 700
670 IF ET THEN 1570
680 GOSUB 110
690 RETURN
700 REM **EXPLOSION SORTIE**
710 SB(X,Y)=SB(X,Y)+PL
720 NE=0
730 A=X
740 B=Y
750 IF DI THEN GOSUB 1710
760 XP=0
770 FOR X=1 TO SI
780 FOR Y=1 TO SI
790 IF ABS(SB(X,Y))<ST(X,Y) THEN 980
800 XP=-1
810 NE=NE+1
820 SB(X,Y)=SB(X,Y)-ST(X,Y)*PL
830 A=X
840 B=Y
850 IF DI THEN GOSUB 1710
860 EX=X
870 EY=Y+1
880 GOSUB 1040
890 EX=X+1
900 EY=Y
910 GOSUB 1040
920 EX=X
930 EY=Y-1
940 GOSUB 1040
950 EX=X-1
960 EY=Y
970 GOSUB 1040
980 NEXT Y
990 NEXT X
1000 ET=(NE>SI*SI)
1010 IF ET THEN 1030
1020 IF XP THEN 760
1030 RETURN

```

```

1040 REM **ENVIRONNEMENT**
1050 SB(EX,EY)=PL*(ABS(SB(EX,EY))+1)
1060 A=EX
1070 B=EY
1080 IF NOT DI THEN 1100
1090 IF ST(EX,EY)>0 THEN GOSUB 1710
1100 RETURN
1110 REM **MOUVEMENT ORDINATEUR**
1120 BE=1000
1130 FOR TX=1 TO SI
1140 FOR TY=1 TO SI
1150 IF RB(TX,TY)>0 THEN 1320
1160 GOSUB 50
1170 X=TX
1180 Y=TY
1190 DI=0
1200 GOSUB 700
1210 IF NOT ET THEN 1250
1220 MX=TX
1230 MY=TY
1240 GOTO 1340
1250 GOSUB 1430
1260 IF EN>BE THEN 1320
1270 IF EN<BE THEN 1290
1280 IF RND>.5 THEN 1320
1290 BE=EN
1300 MX=TX
1310 MY=TY
1320 NEXT TY
1330 NEXT TX

```



```

1340 REM **MOUVEMENT EFFECTIF**
1350 GOSUB 50
1360 X=MX
1370 Y=MY
1380 DI=-1
1390 GOSUB 700
1400 IF ET THEN 1570
1410 GOSUB 110
1420 RETURN
1430 REM **EVALUATION SITUATION**
1440 EN=0
1450 FOR X=1 TO SI
1460 FOR Y=1 TO SI
1470 EN=EN+SB(X,Y)
1480 IF -SB(X,Y)<ST(X,Y)-1 THEN 1540
1490 EN=EN-2
1500 IF SB(X+1,Y)=ST(X+1,Y)-1 THEN EN=EN+10
1510 IF SB(X,Y+1)=ST(X,Y+1)-1 THEN EN=EN+10
1520 IF SB(X-1,Y)=ST(X-1,Y)-1 THEN EN=EN+10
1530 IF SB(X,Y-1)=ST(X,Y-1)-1 THEN EN=EN+10
1540 NEXT Y
1550 NEXT X
1560 RETURN
1570 REM **FIN**
1580 PRINT "L'EXPLOSION FINALE S'EST      PRODUITE. ";
1590 IF PL=1 THEN PRINT "VOUS AVEZ";ELSE PRINT "J'AI";
1600 PRINT " GAGNE":"EN ";NT;"COUPS"
1610 END
1620 REM **PROGRAMME PRINCIPAL**
1630 GOSUB 170
1640 IF CM THEN 1670
1650 PL=1
1660 GOSUB 450
1670 PL=-1
1680 GOSUB 1110
1690 NT=NT+1
1700 GOTO 1650
1710 REM **IMPRESSION SB(A,B)**
1720 IF SB(A,B)<0 THEN C=45 ELSE C=32
1730 CALL HCHAR(2*B+4,3*A+10,C)
1740 CALL HCHAR(2*B+4,3*A+11,ASC(STR$(ABS(SB(A,B)))))
1750 RETURN

```

NEW-YORK, NEW-YORK

Ce jeu original et intéressant vous transporte en hélicoptère très haut au-dessus de New-York et de ses rues. Au début, les rues sont désertes, mais bientôt des voitures apparaissent et il va dépendre de vous qu'elles puissent circuler! Pour cela, vous devez contrôler les feux de signalisation. Lorsqu'une voiture (représentée par un carré) atteint un feu rouge, elle s'arrête. Chaque feu de circulation a un symbole - en pressant la touche sur laquelle figure ce symbole vous faites passer le feu au vert, permettant à la voiture de continuer sa route.


Au début du jeu, vous avez 100 points. Vous gagnez 10 points chaque fois qu'une voiture franchit un carrefour et 25 chaque fois qu'elle traverse d'une traite la ville. En revanche, si deux voitures se tamponnent à un croisement, vous perdez 100 points et bloquez le trafic pendant un moment. Vous perdez aussi des points lorsqu'une voiture doit attendre à un feu. Si vous avez épuisé tous vos points, le jeu s'arrête, sinon, il dure un temps donné. Plus vous avez de points, plus vous méritez vos galons d'agent de la circulation!

```
10 REM TI(ETENDU)-NEW YORK, NEW YORK
30 REM MON RECORD EST 2270 POINTS
40 REM
50 AW=40 :: DIM CX(40),CY(40),CR(40),PX(37),PY(37),P$(37),AX(3),AY(3),FX(9),FY(9)
60 RESTORE :: CALL CLEAR :: CALL CHARSET :: CALL SCREEN(15):: PU=100
70 FOR I=0 TO 40 :: CR(I)=0 :: NEXT I
80 TIME=0 :: TL=0 :: AF=0
90 FOR I=32 TO 44 :: READ X$ :: CALL CHAR(I,X$):: NEXT I
100 DATA "55AA55AA55AA55AA","252A25EAE52A252A","55AA5500FF181818","54A454A757A45
4A4","181818FF00AA55AA"
110 DATA "1818181818181818","000000FFFF000000","181818FFFF181818","FFFFFFFFFFFFFF
FFF"
120 DATA "0F0F0F0F0F0F0F0F","F0F0F0F0F0F0F0F0","00000000FFFFFFFF","FFFFFFFF00000
000"
130 FOR I=48 TO 95 :: CALL CHARPAT(I,X$):: CALL CHAR(48+I,X$):: NEXT I
140 CALL COLOR(1,16,2)! RUES
150 CALL COLOR(2,6,2)! VOITURES
160 FOR I=3 TO 8 :: CALL COLOR(I,2,7):: CALL COLOR(6+I,2,4):: NEXT I ! FEUX
```

```

170 CALL HCHAR(5,1,38,32):: CALL HCHAR(13,5,38,24):: CALL HCHAR(20,1,38,32)
180 CALL VCHAR(1,5,37,24):: CALL VCHAR(5,13,37,8):: CALL VCHAR(5,20,37,16):: CAL
L VCHAR(1,28,37,24)
190 FOR I=1 TO 17 :: READ X,Y,CH :: CALL HCHAR(Y,X,CH):: NEXT I
200 DATA 13,4,34,20,4,34,5,5,39,13,5,39,20,5,39,28,5,39
210 DATA 4,13,35,5,13,39,13,13,39,20,13,39,28,13,39,29,13,33,13,14,36
220 DATA 5,20,39,20,20,39,28,20,39,20,21,36
230 FOR I=0 TO 37 :: READ PX(I),PY(I),P$(I):: CALL HCHAR(PY(I),PX(I),ASC(P$(I)))
:: NEXT I
240 DATA 4,3,A,27,3,B,7,4,C,15,4,D,22,4,E,30,4,F,3,6,6,11,6,H,18,6,I,26,6,J
250 DATA 6,7,K,14,7,L,21,7,M,29,7,N,4,11,0,12,11,P,19,11,Q,27,11,R
260 DATA 7,12,S,15,12,T,22,12,U,11,14,V,18,14,W,26,14,X,6,15,Y,21,15,Z,29,15,0
270 DATA 4,18,1,19,18,2,27,18,3,7,19,4,22,19,5,30,19,6,3,21,7,18,21,8,26,21,9,6,
22,<,29,22,>
280 X$="NEW_YORK" :: DISPLAY AT(2,6):X$;
290 FOR I=1 TO 8 :: CALL HCHAR(2,17+I,ASC(SEG$(X$,I,1))+48):: NEXT I
300 DISPLAY AT(16,6):"VOUS AVEZ";:: DISPLAY AT(17,6):"100 POINTS";
310 RANDOMIZE
320 DEF FNR(X)=INT(RND*X)
330 AX(0)=1 :: AX(1)=-1 :: AY(2)=1 :: AY(3)=-1
340 FOR I=0 TO AW :: R=CR(I):: GOSUB 950 :: IF R=0 THEN 440
350 X=CX(I):: Y=CY(I)
360 IF R AND 1 THEN X=X+1 :: F=43 :: 00=44 :: 0=38 :: SX=0 :: SY=1 :: IF X>32 TH
EN 480
370 IF R AND 2 THEN X=X-1 :: F=44 :: 00=43 :: 0=38 :: SX=0 :: SY=-1 :: IF X<1 TH
EN 480
380 IF R AND 4 THEN Y=Y+1 :: F=42 :: 00=41 :: 0=37 :: SX=-1 :: SY=0 :: IF Y>24 T
HEN 480
390 IF R AND 8 THEN Y=Y-1 :: F=41 :: 00=42 :: 0=37 :: SX=1 :: SY=0 :: IF Y<1 THE
N 480
400 IF R AND 16 THEN 450
410 IF R AND 32 THEN 560
420 IF R AND 64 THEN 0=39 :: R=R-64 :: CR(I)=R :: PU=PU+10
430 GOTO 490
440 NEXT I :: GOTO 660
450 CALL GCHAR(CY(I)+SY,CX(I)+SX,LIGHT)
460 IF LIGHT>95 THEN CR(I)=(R OR 32)AND NOT 16 :: GOTO 490
470 PU=PU-1 :: GOTO 440
480 PU=PU+25 :: PN=PN+1 :: CR(I)=0 :: GOTO 550
490 CALL GCHAR(Y,X,PS):: IF PS=F OR PS=40 THEN PU=PU-1 :: GOTO 440
500 CALL GCHAR(Y+SY,X+SX,A):: IF A<>32 THEN CR(I)=R OR 16
510 IF PS=00 THEN F=40
520 CALL GCHAR(CY(I),CX(I),V):: IF V=40 THEN 0=00
530 IF V=32 THEN CR(I)=0 :: A0=A0+1 :: PU=PU-100 :: GOTO 440

```



```

540 CALL HCHAR(Y,X,F)
550 CALL HCHAR(CY(I),CX(I),0):: CX(I)=X :: CY(I)=Y :: GOTO 440
560 CALL GCHAR(Y,X,A):: IF A=F THEN 440
570 IF A<>39 THEN 630
580 VB=0 :: FOR J=0 TO 3 :: A=2^J :: CALL GCHAR(Y+AY(J),X+AX(J),B):: IF NOT(B=3B
OR B=37)THEN VB=VB OR A
590 NEXT J
600 B=2^FNR(4):: IF (B AND VB)=0 THEN CR(I)=B OR 64 :: GOTO 540
610 IF VB=15 THEN 440
620 GOTO 600
630 AO=AO+1 :: CR(I)=0 :: IF A<>32 THEN FX(AF)=X :: FY(AF)=Y :: AF=AF+1 :: PU=PU
-100
640 F=32 :: CALL SOUND(4000,-6,0)
650 GOTO 540
660 IF CR(TL)=0 THEN 690
670 TL=TL+1 :: IF TL>AW THEN TL=0
680 GOTO 780
690 A=FNR(4):: CR(TL)=2^A :: ON A+1 GOTO 700,710,720,730
700 CX(TL)=1 :: GOTO 740
710 CX(TL)=32 :: GOTO 740
720 CY(TL)=1 :: GOTO 760
730 CY(TL)=24 :: GOTO 760
740 IF FNR(2)=1 THEN CY(TL)=5 :: GOTO 780
750 CY(TL)=20 :: GOTO 780
760 IF FNR(2)=1 THEN CX(TL)=5 :: GOTO 780
770 CX(TL)=28 :: GOTO 780
780 REM GESTION DU TEMPS
790 IF PU<0 THEN PU=0

```



```

800 DISPLAY AT(17,5):USING "####":PU;
810 IF PU=0 THEN 880
820 TIME=TIME+1 :: IF TIME>100 THEN 880
830 IF FNR(25)>0 OR AF=0 THEN 870
840 CALL HCHAR(FY(0),FX(0),39)
850 FOR I=0 TO 8 :: FY(I)=FY(I+1):: FX(I)=FX(I+1):: NEXT I
860 AF=AF-1
870 GOTO 340
880 IF PU=0 THEN CALL SOUND(4000,-1,0):: GOTO 900
890 FOR I=1 TO 5 :: CALL SOUND(1000,698,0):: CALL SOUND(1000,784,0):: CALL SOUND
(1000,880,0):: NEXT I
900 DISPLAY AT(2,6):"ON_REJOU?_0;N 0 ";;: ACCEPT AT(2,22)SIZE(-1)VALIDATE("ON"
):A$
910 IF A$="N" THEN 930
920 GOTO 60
930 CALL CLEAR :: CALL CHARSET
940 PRINT "VOUS AVEZ";PU;"POINTS" :: END
950 REM TOUCHE ENFORCEE?
960 CALL KEY(0,LIGHT,CH)
970 IF CH<>1 THEN RETURN
980 IF LIGHT>64 AND LIGHT<91 THEN NR=LIGHT-65 :: GOTO 1030
990 IF LIGHT>47 AND LIGHT<58 THEN NR=LIGHT-22 :: GOTO 1030
1000 IF LIGHT=44 THEN NR=36 :: GOTO 1030
1010 IF LIGHT=46 THEN NR=37 :: GOTO 1030
1020 RETURN
1030 CALL GCHAR(PY(NR),PX(NR),COLLIGHT)
1040 IF COLLIGHT>95 THEN NEWCH=COLLIGHT-48 :: GOTO 1060
1050 NEWCH=COLLIGHT+48
1060 CALL HCHAR(PY(NR),PX(NR),NEWCH)
1070 CALL SOUND(-200,1047,0)
1080 RETURN

```

LA CLE

Chercher une clé qu'on a perdue est une expérience éprouvante en soi, mais lorsqu'il vous faut la trouver le plus rapidement possible et que, de surcroît, elle est cachée à l'intérieur d'un ordinateur, toute l'affaire devient franchement exaspérante, mais ô combien excitante!

L'ordinateur génère une suite de 20 zéro et un.

0 1 1 0 1 0 1 0 1 1 1 0 1 0 1 0 0 1 0 1

Cette suite peut être décalée cycliquement, c'est-à-dire que les chiffres de l'extrême-droite passent à gauche. On fait cela trois fois et l'on additionne les nombres obtenus. Exemple:

LIGNE 1 →	00	1	0	1	0	1	1	1	0	1	0	1	0	1	1	0	1	0	1
LIGNE 2 →	0	1	1	1	0	1	0	1	0	1	1	0	1	0	1	00	1	0	1
LIGNE 3 →	0	1	00	1	0	1	0	1	1	1	0	1	0	1	0	1	1	0	1
SOMME →	022	1	2	1	2	2	2	3	0	3	0	3	1	1	3	0	3		

Mais cette somme vous est cachée: c'est la clé que vous devez trouver. Ce que vous voyez, ce sont les trois rangées de nombres, chaque rangée ayant été à nouveau décalée. Si, par exemple, la ligne du haut a été décalée de 18 positions, celle du milieu de 16 positions et celle du bas de 12 positions, vous verrez:

1	0	1	0	1	1	1	0	1	0	1	0	1	1	0	1	0	1	0	0
0	1	0	1	0	1	1	0	1	0	1	0	0	1	0	1	0	1	1	1
1	1	1	0	1	0	1	0	1	1	0	1	0	1	0	0	1	0	1	0
<hr/>																			
2	0	0	0	0	1	1	2	1	1	1	1	2	3	3	1	0	1	2	2

On n'affiche pas la somme des trois lignes mais la différence entre cette somme et la somme "cachée". C'est ainsi que pour la première colonne la valeur cachée est 0, la nouvelle somme est 2, on affiche donc 2; pour la seconde colonne la somme cachée est 2 de même que la nouvelle somme, on affiche donc 0. Vous devez maintenant décaler vous-même les lignes de telle sorte qu'elles finissent par être dans la même position que la clé, et donc que la différence soit à 0. Les lignes sont numérotées de 1 à 3 (celle du sommet porte le numéro 1). Si vous tapez

```
LIGNE = 2
MOUVEMENTS = 1
```


La ligne du milieu sera décalée d'une position vers la droite. Combien de coups vous faudra-t-il pour trouver la clé? On peut y arriver en 10 coups.

```
10 REM TI-CLE
20 DIM T(4,20)
30 A$="10100101011101010110"
40 GOTO 520
50 REM TIRAGE ALEATOIRE***
60 RANDOMIZE
70 FOR J=1 TO 3
80 R=INT(RND*19+1)
90 FOR K=1 TO 20
100 DIS=K+R
110 IF DIS<21 THEN 130
120 DIS=DIS-20
130 T(J,K)=VAL(SEG$(A$,DIS,1))
140 NEXT K
150 NEXT J
```

```

160 RETURN
170 REM CALCUL CLE*****
180 FOR K=1 TO 20
190 T(4,K)=T(1,K)+T(2,K)+T(3,K)
200 NEXT K
210 RETURN
220 REM ECRAN*****
230 CALL CLEAR
240 TEL=0
250 FOR K=1 TO 20
260 CALL HCHAR(10,K+2,ASC(STR$(T(1,K))))
270 CALL HCHAR(11,K+2,ASC(STR$(T(2,K))))
280 CALL HCHAR(12,K+2,ASC(STR$(T(3,K))))
290 V=T(4,K)-T(3,K)-T(2,K)-T(1,K)
300 V=ABS(V)
310 IF V<>0 THEN 330
320 TEL=TEL+1
330 CALL HCHAR(14,K+2,ASC(STR$(V)))
340 NEXT K
350 RETURN
360 REM ENTREE*****
370 INPUT "LIGNE(1-3):":J
380 IF (J<1)+(J>3)<>0 THEN 370
390 INPUT "MOUVEMENTS(1-19):":S
400 IF (S<1)+(S>19)<>0 THEN 390
410 NT=NT+1
420 RETURN

```



```

430 REM DECALAGE LIGNE*****
440 FOR I=1 TO 5
450 H=T(J,20)
460 FOR K=19 TO 1 STEP -1
470 T(J,K+1)=T(J,K)
480 NEXT K
490 T(J,1)=H
500 NEXT I
510 RETURN
520 REM PROGRAMME PRINCIPAL**
530 GOSUB 50
540 GOSUB 170
550 GOSUB 50
560 GOSUB 220
570 IF TEL=20 THEN 610
580 GOSUB 360
590 GOSUB 430
600 GOTO 560
610 PRINT "REUSSI EN ";NT;" COUPS"
620 END


```

LA BOÎTE NOIRE


Armé d'un rayon laser vous devez retrouver une poignée d'atomes cachés à l'intérieur d'une grande boîte noire.

La boîte est faite de 3 fois 8 cubes. Des atomes peuvent se cacher dans n'importe quel cube, sauf dans les cubes qui se trouvent sur les bords de la boîte, de plus, il y a toujours 5 atomes dans la boîte. Ces atomes renverront ou dévieront les rayons laser selon les règles suivantes.


- un rayon qui frappe un atome est purement et simplement réfléchi




- un rayon qui traverse un cube directement adjacent à un atome est dévié à angle droit.



- un rayon qui traverse un cube diagonalement adjacent à un atome est renvoyé dans une direction qu'on obtient en ajoutant deux réflexions à angles droits.



La boîte est numérotée de la façon suivante:



L'ordinateur vous demande d'entrer P1, P2 et P3, les coordonnées de l'endroit où vous voulez que le rayon entre dans la boîte. Cet endroit se situant évidemment à la surface, au moins une des coordonnées doit être 1 ou 8. Une fois que vous avez entré ces coordonnées, l'ordinateur vous dira où le rayon est ressorti.

Rappelez-vous qu'un rayon peut être réfléchi par plus d'un atome. Si 2 atomes se trouvent exactement à côté l'un de l'autre, la direction finale du rayon est déterminée par la somme des effets de chaque atome. Après avoir envoyé "quelques" tirs, vous devriez être en mesure de trouver l'emplacement des atomes. Mais chez vous, combien font "quelques"?

```
10 REM TI-BOITE NOIRE
20 RANDOMIZE
30 FOR T=1 TO 5
40 X=INT(RND*6+2)
42 Y=INT(RND*6+2)
44 Z=INT(RND*6+2)
46 IF M(X,Y,Z)=1 THEN 40
48 M(X,Y,Z)=1
50 NEXT T
60 PRINT "ENTREE DU TIR"
70 INPUT "P1=":P1
80 INPUT "P2=":P2
90 INPUT "P3=":P3
100 S1=(P1=8)-(P1=1)
110 S2=(P2=8)-(P2=1)
120 S3=(P3=8)-(P3=1)
130 IF (P1=8)+(P1=1)+(P2=8)+(P2=1)+(P3=8)+(P3=1)=-1 THEN 160
140 PRINT "ENTREE ERRONEE"
150 GOTO 60
160 FOR A1=-1 TO 1
170 FOR A2=-1 TO 1
180 FOR A3=-1 TO 1
190 IF M(P1+A1,P2+A2,P3+A3)<>1 THEN 230
200 S1=S1-A1
210 S2=S2-A2
220 S3=S3-A3
230 NEXT A3
240 NEXT A2
250 NEXT A1
```



```

260 S1=SGN(S1)
270 S2=SGN(S2)
280 S3=SGN(S3)
290 P1=P1+S1
300 P2=P2+S2
310 P3=P3+S3
320 IF (P1=0)+(P1=9)+(P2=0)+(P2=9)+(P3=0)+(P3=9)=0 THEN 160
330 PRINT "RESULTAT:";P1-S1;P2-S2;P3-S3
340 INPUT "TIR OU POSITION ATOME?      (T OU P) ":AW$
350 IF AW$="T" THEN 60
360 IF AW$="P" THEN 380
370 END
380 PRINT "TAPEZ LES COORDONNEES"
390 INPUT "P1=":P1
400 INPUT "P2=":P2
410 INPUT "P3=":P3
420 IF M(P1,P2,P3)=1 THEN 450
430 PRINT "FAUX"
440 GOTO 340
450 PRINT "BRAVO"
460 GOTO 340


```

CHASSE AU TRESOR

Avez-vous déjà rêvé d'aller à la chasse au trésor? De parcourir un pays sauvage et hostile, chassant et pêchant pour survivre, dormant à la dure, jusqu'à atteindre des contrées lointaines et interdites où se cache votre trésor étincelant de mille feux.

Si c'est le cas, ce jeu est fait pour vous.

On va vous transporter en haut des Alpes et vous donner une carte indiquant l'emplacement des trésors et leur valeur. Pour atteindre ces trésors vous devrez cheminer le long d'étroits sentiers tortueux en utilisant les touches de déplacement du curseur de votre ordinateur - un faux pas et c'est la mort. Vous découvrirez ces trésors, simplement en marchant dessus.



Mais voilà, ce n'est pas si facile. Vous ne pensiez tout de même pas que ce serait si simple, non? Les trésors que vous cherchez sont gardés avec vigilance par quelques oiseaux terrifiants et féroces qui vous atta-

queront si vous ne faites pas attention. S'ils captent votre regard, vous serez paralysés par la peur. Tout ce que vous pouvez faire pour essayer de les arrêter est de taper sur la barre d'espacement.

Si vous vous sentez encore de taille à relever le défi, commencez votre recherche dans le coin gauche de la carte. Ramassez le plus de trésors possibles et emportez-les vers la sortie en haut à droite. Cela vous donnera un bonus et une possibilité supplémentaire de voyager le long des chemins en ramassant des trésors. En réalité, vous pouvez faire autant de voyages que vous le voulez à la condition que vous puissiez fuir ensuite. Car malheureusement chaque fois que vous repartez vous dérangez un peu plus ces oiseaux effroyables, qui deviennent de plus en plus agressifs.

Les risques sont élevés, mais les récompenses aussi, alors rassemblez votre courage et en avant!

```
10 REM TI-CHASSE AU TRESOR
20 DIM B(16,21)
30 GOTO 680
40 REM ECRAN*****
50 CALL CHAR(136,"18183C5A993C2466")
60 CALL CHAR(95,"00183C7E7E3C1000")
70 CALL COLOR(14,2,1)
80 CALL CHAR(120,"FFFFFFFFFFFFFF")
90 CALL COLOR(12,2,2)
100 CALL CLEAR
110 CALL COLOR(8,5,16)
120 DATA 30,30,30,30,30,30,30,30,30,30,30,30,30,30,30,30,31,30
130 DATA 30,31,31,31,51,30,50,30,30,31,31,31,31,31,54,30,31,31,31,30
140 DATA 30,53,30,31,30,30,31,31,31,31,30,30,52,30,30,31,31,30,30,30
150 DATA 30,30,30,31,31,31,30,31,30,31,31,30,30,53,30,31,30,30,55,30
160 DATA 30,31,31,31,30,51,30,31,30,30,31,31,31,31,30,31,31,31,31,30
170 DATA 30,31,30,31,30,30,31,31,31,30,31,30,31,30,30,31,30,30,30,30
180 DATA 30,31,30,31,31,30,31,30,31,30,31,30,53,30,31,31,30,31,57,30
190 DATA 30,31,30,30,31,31,31,30,51,30,31,30,30,30,31,30,30,31,30,30
200 DATA 30,31,31,31,30,30,31,30,30,30,31,30,52,30,31,31,30,31,56,30
210 DATA 30,49,30,31,30,51,31,30,31,31,31,31,31,30,30,31,30,31,30,30
220 DATA 30,30,30,31,30,30,30,49,31,30,30,30,30,51,30,31,30,31,31,30
230 DATA 30,31,31,31,31,49,30,30,31,30,31,31,31,31,30,31,30,30,31,30
240 DATA 30,31,30,31,30,30,51,30,31,30,31,30,31,30,30,31,31,31,31,30
250 DATA 30,31,30,31,31,31,31,30,31,31,31,30,31,31,31,31,30,30,55,30
260 DATA 30,31,30,30,30,30,30,30,30,30,30,30,30,30,30,30,30,30,30,30
270 RESTORE
```

```

280 FOR C1=1 TO 15
290 FOR C2=1 TO 20
300 READ B(C1,C2)
310 IF B(C1,C2)<>30 THEN 330
320 B(C1,C2)=120
330 CALL HCHAR(C1+5,C2+5,B(C1,C2))
340 NEXT C2
350 NEXT C1
360 RETURN
370 REM MOUVEMENTS JOUEUR**
380 CALL KEY(3,K,S)
390 X1=X+(K=69)-(K=88)
400 Y1=Y+(K=83)-(K=68)
410 IF (B(X1,Y1)=120)+(X1<1)+(X1>15)+(Y1<1)+(Y1>20)<>0 THEN 460
420 CALL HCHAR(X+5,Y+5,B(X,Y))
430 X=X1
440 Y=Y1
450 CALL HCHAR(X+5,Y+5,136)
460 RETURN
470 REM ATTAQUE OISEAUX****
480 IF RND>Q THEN 610

```



```

490 FOR CC=25 TO 6 STEP -1
500 CALL HCHAR(X+5,CC,95)
510 CALL KEY(3,K,S)
520 IF K=32 THEN 550
530 IF CC<>Y+5 THEN 560
540 BS=-1
550 CC=1
560 NEXT CC
570 FOR CC=6 TO 25
580 CALL HCHAR(X+5,CC,B(X,CC-5))
590 NEXT CC
600 CALL HCHAR(X+5,Y+5,136)
610 RETURN
620 REM INITIALISATION*****
630 NG=NG+1
640 X=15
650 Y=2
660 Q=Q+.1
670 RETURN
680 REM PROGRAMME PRINCIPAL***
690 GOSUB 40
700 GOSUB 620
710 GOSUB 370
720 IF (X=1)+(Y=19)=-2 THEN 690
730 IF B(X,Y)=31 THEN 760
740 SCORE=SCORE+NG*(B(X,Y)-48)
750 B(X,Y)=31
760 GOSUB 470
770 IF BS=0 THEN 710
780 PRINT "VOUS AVEZ RAMASSE";SCORE;"FRANCS"
790 END

```

I.T. OU LE VOYAGE AU CENTRE DE LA TERRE

Dans cette folle aventure vous allez plonger dans les entrailles de la terre et rencontrer une très étrange créature qui y vit. Son nom: I.T. bien sûr, comme Intra-Terrestre.

Si vous n'avez jamais joué à un jeu d'aventure auparavant, ne vous souciez pas des règles, il n'y en a pas. Vous devez tout faire par vous-même au cours de ce voyage au centre de la Terre, accomplir des travaux extraordinaires et affronter des dangers terrifiants. Si vous ne vous en sortez pas, tapez HELP et regardez ce que l'ordinateur vous répond. Quand vous aurez tapé RUN, il vous faudra attendre un bon moment avant de voir quelque chose sur l'écran. Dans presque tous les cas, vous pouvez ne taper que la première lettre du mot choisi, par exemple N pour nord.

Si cela vous tente, prenez une bonne assurance-vie, tapez la commande RUN, et à vous de jouer!

```
10 REM TI(ETENDU)-I.T.
20 RANDOMIZE
30 DIM K(36,6),R$(36),H$(15),VP(12),V$(12),B(12),PR(6,3)
40 GOSUB 2810
50 IHD=8 :: NKAM=0
60 GOTO 80
70 GOSUB 130
80 GOSUB 390
90 GOSUB 290
100 IF CWD<5 THEN 70
110 PRINT "VOUS ETES MORT"
120 END
130 NKAM=0
140 IHD=0
150 INPUT "QUE VOULEZ-VOUS FAIRE?":C1$
160 LNG=LEN(C1$)
170 FOR I=1 TO 15
180 IF SEG$(H$(I),1,LNG)=C1$ THEN IHD=I :: I=16
```

```

190 NEXT I
200 IF IHD>6 THEN RETURN
210 IF IHD=0 THEN 260
220 NKAM=K(IKM,IHD)
230 IF NKAM<>0 THEN RETURN
240 PRINT "IMPOSSIBLE"
250 GOTO 130
260 PRINT "JE NE COMPRENDS PAS"
270 GOTO 130
280 RETURN
290 PRINT "VOUS ETES DANS LE ";R$(IKM)
300 IF IKM>24 OR BLL=1 THEN 330
310 PRINT "ON N'Y VOIT RIEN"
320 GOTO 340
330 IF IVW<>0 THEN PRINT "IL Y A UN ";V$(IVW)
340 FOR I=1 TO 6
350 IF K(IKM,I)<>0 THEN PRINT "VOUS POUVEZ ALLER ";H$(I)
360 NEXT I
370 PRINT
380 RETURN
390 IF NKAM<>0 THEN IKM=NKAM
400 IVW=0
410 FOR I=1 TO 12
420 IF VP(I)=IKM THEN IVW=I :: I=12
430 NEXT I
440 IPS=0
450 FOR I=1 TO 6
460 IF PR(I,1)=IKM THEN IPS=I :: I=6
470 NEXT I
480 IF IKM<12 AND CRI<>8 THEN GOSUB 2670
490 CLL=CLL+BLL
500 IF IHD<7 THEN 530
510 ON IHD-6 GOSUB 560,760,810,910,1130,1220,1340,1420,1500
520 REM
530 IF IPS=0 THEN RETURN
540 ON IPS GOSUB 1620,1840,2080,2180,2290,2470
550 RETURN
560 IF IPS=0 OR IPS>4 THEN PRINT "LE TEMPS EST LA SEULE CHOSE QUE VOUS POUVEZ TU
ER." :: RETURN
570 IVWB=7
580 GOSUB 2610
590 IF IBZ=0 THEN PRINT "N'ATTAQUEZ PAS SANS EPEE" :: RETURN
600 PRINT "VOUS VOULEZ TUER LE ";P$(IPS)
610 PRINT "VOUS L'ATTAQUEZ AVEC VOTRE GLAIVE"

```

```

620 IF RND<.5 THEN PRINT "MAIS VOUS LE MANQUEZ" :: GOTO 650
630 PRINT "ET VOUS LUI ASSENEZ UN COUP TERRIBLE"
640 PR(IPS,3)=PR(IPS,3)-1
650 IF RND>.2 THEN 700
660 PRINT "VOUS POUVEZ L'AVOIR MAIS FAITES VITE"
670 INPUT "VOUS VOULEZ?(O OU N)":Y$
680 IF Y$<>"O" THEN 700
690 PRINT "VOUS LEVEZ VOTRE GLAIVE";: GOTO 620
700 IF PR(IPS,3)>1 THEN RETURN
710 IF PR(IPS,3)=1 THEN PRINT "IL EST MECHAMMENT BLESSE" :: RETURN
720 PRINT "VOUS L'AVEZ TUE"
730 PR(IPS,1)=0 :: IVW=8+IPS
740 VP(IVW)=IKM :: IPS=0
750 RETURN
760 PRINT "COMMANDES DISPONIBLES:"
770 FOR I=1 TO 15
780 PRINT H$(I)
790 NEXT I
800 RETURN
810 IF IVW<>0 THEN 840
820 PRINT "IL N'Y A RIEN A PRENDRE"
830 RETURN
840 IF BLL=0 AND IKM<25 THEN PRINT "NE PRENEZ QUE CE QUE VOUS VOYEZ." :: RETURN
850 PRINT "VOUS PRENEZ LE ";V$(IVW)
860 B(IBMAX)=IVW
870 IBMAX=IBMAX+1
880 VP(IVW)=0
890 IVW=0
900 RETURN
910 INPUT "QUE VOULEZ-VOUS POSER?":D$
920 LNG=LEN(D$)
930 IVWB=0
940 FOR I=1 TO 12
950 IF D$=SEG$(V$(I),1,LNG) THEN IVWB=I :: I=12
960 NEXT I
970 IF IVWB<>0 THEN 1000
980 PRINT "JE NE COMPRENDS PAS"
990 RETURN
1000 GOSUB 2610
1010 IF IBZ<>0 THEN 1040
1020 PRINT "IMPOSSIBLE DE POSER"
1030 RETURN
1040 PRINT "VOUS AVEZ POSE LE ";V$(B(IBZ))

```



```


1050 IVW=B(IBZ)
1060 VP(B(IBZ))=IKM
1070 B(IBZ)=B(IBMAX-1)
1080 IBMAX=IBMAX-1
1090 IF IKM<>25 THEN RETURN
1100 INPUT "VOULEZ-VOUS POSER AUTRE      CHOSE?(O OU N)":Y$
1110 IF Y$="O" THEN 910
1120 RETURN
1130 IF IBMAX<>1 THEN 1160
1140 PRINT "VOUS N'AVEZ RIEN"
1150 GOTO 1200
1160 PRINT "VOUS AVEZ:"
1170 FOR I=1 TO IBMAX-1
1180 PRINT V$(B(I))
1190 NEXT I
1200 PRINT "VOUS POUVEZ ETRE BLESSE ";5-CWD;"FOIS AVANT DE MOURIR"
1210 RETURN
1220 IVWB=6
1230 GOSUB 2610
1240 IF IBZ<>0 THEN 1270
1250 PRINT "IMPOSSIBLE SANS PANSEMENT"
1260 RETURN
1270 PRINT "J'UTILISE LES PANSEMENTS"
1280 CALL SOUND(100,1000,0)
1290 PRINT "CA TIENDRA BIEN UN MOMENT"
1300 CWD=0
1310 B(IBZ)=B(IBMAX-1)
1320 IBMAX=IBMAX-1
1330 RETURN
1340 IVWB=1
1350 GOSUB 2610
1360 IF IBZ<>0 THEN 1390
1370 PRINT "IMPOSSIBLE SANS PROJECTEUR"
1380 RETURN
1390 PRINT "LE PROJECTEUR EST ALLUME"
1400 BLL=1
1410 RETURN
1420 IVWB=1
1430 GOSUB 2610
1440 IF IBZ<>0 THEN 1470
1450 PRINT "VOUS N'AVEZ MEME PAS DE PROJECTEUR"
1460 RETURN
1470 PRINT "LE PROJECTEUR EST ETEINT"
1480 BLL=0

```

```

1490 RETURN
1500 IVWB=4
1510 GOSUB 2610
1520 IF IBZ<>0 THEN 1550
1530 PRINT "RIEN A LIRE"
1540 RETURN
1550 IF BLL=1 OR IKM>24 THEN 1580
1560 PRINT "IL FAIT TROP SOMBRE POUR LIRE"
1570 RETURN
1580 T$="JM!Z!B!VOF!SFDFUUF!EF!QFUJUTHBUFBVY!EBOT!DF!MJWSF" :: GOSUB 3520
1590 T$="QSFOESF!VO!WJFVY!CPVD!FU!VO!TBD!B!GBSJO!MFT!KFUFS!EBOT!VO!DIBVESPO!FU!
BUUFOESF" :: GOSUB 3520
1600 PRINT "C'EST TOUT"
1610 RETURN
1620 IF IVW=3 THEN PR(1,2)=3 :: VP(IVW)=0 :: IVW=0
1630 ON PR(1,2)GOTO 1640,1690,1720,1790,1820
1640 PRINT "I.T.(L'INTRA-TERRESTRE) EST LA."
1650 PRINT "IL VEUT PARLER"
1660 PRINT "IL PARLE MAIS VOUS NE LE      COMPRENEZ PAS."
1670 PR(1,2)=2
1680 RETURN
1690 T$="J/U!/FYQMJRVE!RV(JM!WVU!!!!CPJSF" :: GOSUB 3520
1700 T$="JM!B!M(BJS!EFTFTQFSF" :: GOSUB 3520
1710 RETURN
1720 T$="J/U!/BWBJU!TPJG" :: GOSUB 3520
1730 T$="JM!EJU!T(JM!WPVT!QMBJU!!!!EFTDFOEFC!EBOT!MFT!!!!!!TPVUFSSBJOT///U
SPVWFS!M(PCVT!B!CFSPOT" :: GOSUB 3520
1740 T$="JM!WB!FYQMPFST///VO!NPOTUSF!MB!HBSEF" :: GOSUB 3520

```



```

1750 T$="WPVT!TFVM!QPWF!TBVWF!MB!!UFSSF" :: GOSUB 3520
1760 PRINT "SOUDAIN, I.T. S'EFFONDRE"
1770 PR(1,2)=4
1780 RETURN
1790 PRINT "I.T. EST DANS LE COMA"
1800 PR(1,2)=5
1810 RETURN
1820 PRINT "I.T. EST ICI":"DANS LE COMA"
1830 RETURN
1840 IF IVW=8 THEN PR(2,2)=5 :: VP(IVW)=0 :: IVW=0
1850 ON PR(2,2)GOTO 1860,1900,1930,1960,2000,2060
1860 PRINT "IL Y A UN ENORME MONSTRE."
1870 PRINT "LE MONSTRE CRIE: ES-TU UN GATEAU ?"
1880 PR(2,2)=2
1890 RETURN
1900 PRINT "LE MONSTRE CRIE DE PLUS EN PLUS FORT: ES-TU UN GATEAU ?"
1910 PR(2,2)=3
1920 RETURN
1930 PRINT "LE MONSTRE CONTINUE DE CRIER ET DEVIENT AGRESSIF"
1940 PR(2,2)=4
1950 RETURN
1960 PRINT "LE MONSTRE VOUS DONNE UN COUP TERRIBLE. VOUS VOYEZ 36CHANDELLES."
1970 CWD=CWD+1
1980 PR(2,2)=INT(RND*3)+2
1990 RETURN
2000 PRINT "LE MONSTRE COMMENCE A MANGER."
2010 PRINT "IL S'ENDORT."
2020 CALL SOUND(100,110,0)
2030 K(16,6)=1
2040 PR(2,2)=6
2050 RETURN
2060 PRINT "LE MONSTRE DORT"
2070 RETURN
2080 PRINT "IL Y A UN SERPENT"
2090 IF RND<.4 THEN RETURN
2100 IF IBMAX=1 OR RND<.5 THEN 2140
2110 IBMAX=IBMAX-1
2120 VP(B(IBMAX))=13+INT(RND*12)
2130 PRINT "LE SERPENT PREND QUELQUE CHOSE"
2140 PRINT "ET S'ENFUIT"
2150 PR(3,1)=PR(3,1)+3
2160 IF PR(3,1)>24 THEN PR(3,1)=PR(3,1)-8
2170 RETURN
2180 ON PR(4,2)GOTO 2190,2220,2260

```



```

2190 PRINT "IL Y A UN VIEUX BOUC. IL A L'AIR DE VOUS TROUVER      IMPORTUN"
2200 PR(4,2)=2
2210 RETURN
2220 PRINT "LE VIEUX BOUC VOUS ATTAQUE ET VOUS MORD VIOLEMMENT"
2230 CWD=CWD+1
2240 PR(4,2)=3
2250 RETURN
2260 PRINT "LE VIEUX BOUC REPASSE A      L'ATTAQUE"
2270 PR(4,2)=2+INT(RND*2)
2280 RETURN
2290 PRINT "L'OBUS A ZERONS EST LA"
2300 PRINT "ON VOIT TROIS FILS; UN      GRIS(G), UN ROUGE(R) ET UN JAUNE(J)"
2310 PRINT "VOUS DEVEZ EN DECONNECTER  DEUX"
2320 INPUT "QUEL EST LE PREMIER: ":X$
2330 INPUT "ET LE SECOND: ":Y$
2340 CBO=0 :: C$="RJG"
2350 FOR I=1 TO 3
2360 IF X$=SEG$(C$,I,1)THEN CBO=CBO+1
2370 IF Y$=SEG$(C$,I,1)THEN CBO=CBO+1
2380 NEXT I
2390 IF CBO<2 THEN PRINT "ATTENTION, ENTREE ERRONEE " :: GOTO 2300

```

```

2400 IF (X$&Y$)="GR" OR(X$&Y$)="RG" THEN 2440
2410 CALL CLEAR
2420 PRINT "CE N'ETAIT PAS LES BONS. LA TERRE A SAUTE"
2430 END
2440 CALL CLEAR
2450 PRINT "FELICITATIONS !,VOUS AVEZ  REUSSI OU LES AUTRES AVAIENTECHEUE"
2460 END
2470 IF IVW=0 THEN 2520
2480 PR(6,2)=2
2490 IF VP(2)=IKM AND VP(12)=IKM THEN IVW=8 :: VP(IVW)=IKM :: VP(2)=0 :: VP(12)=
0 :: GOTO 2520
2500 VP(IVW)=0
2510 IVW=0
2520 ON PR(6,2)GOTO 2530,2550,2580
2530 PRINT "IL Y A UN GROS CHAUDRON POSESUR UN GRAND FEU"
2540 RETURN
2550 PRINT "UNE ODEUR FORTE VOUS SAISIT"
2560 PR(6,2)=3
2570 RETURN
2580 PRINT "TOUT EST CALME. MEME LA  TERRIBLE ODEUR A DISPARU"
2590 PR(6,2)=1
2600 RETURN
2610 REM
2620 IBZ=0
2630 FOR I=1 TO IBMAX-1
2640 IF B(I)=IVWB THEN IBZ=I :: I=IBMAX-1
2650 NEXT I
2660 RETURN
2670 REM
2680 CRO=CRO+1
2690 IF IKM=D(CRO)THEN CRI=CRI+1
2700 IF CRO<8 THEN RETURN
2710 IF CRI=8 THEN 2770
2720 PRINT "IL ARRIVE DES CHOSES  ETRANGES.IL FAIT NOIR"
2730 PRINT "VOUS ETES INCONSCIENT  QUELQUES INSTANTS"
2740 CRO=1 :: CRI=1
2750 IKM=1
2760 RETURN
2770 PRINT "VOUS ENTENDEZ UN SON ETRANGEcomme SI ON TIRAIT QUELQUE  CHOSE"
2780 PRINT "MAINTENANT CA S'EST ARRETE"
2790 K(2,1)=3
2800 RETURN
2810 REM
2820 REM

```



```
2830 FOR I=1 TO 36
2840 K(I,1)=I+1 :: K(I,2)=I-1
2850 K(I,3)=I+4 :: K(I,4)=I-4
2860 NEXT I
2870 FOR I=0 TO 24 STEP 12
2880 FOR J=1 TO 9 STEP 4
2890 K(I+J+3,1)=0 :: K(I+J,2)=0
2900 NEXT J
2910 FOR J=1 TO 4
2920 K(I+J+8,3)=0 :: K(I+J,4)=0
2930 NEXT J
2940 NEXT I
2950 K(1,5)=16 :: K(7,5)=15
2960 K(32,6)=13 :: K(13,5)=32
2970 K(35,6)=18 :: K(18,5)=35
2980 FOR I=1 TO 15
2990 READ IKM :: READ IHD
3000 K(IKM,IHD)=0
3010 NEXT I
3020 REM
3030 IKM=36
3040 IBZ=1 :: IBMAX=1
```

```


3050 CKM=0
3060 CLL=0
3070 BLL=0
3080 CWD=0
3090 CRO=0 :: CRI=0
3100 REM
3110 FOR I=1 TO 15
3120 READ H$(I)
3130 NEXT I
3140 IN=-1
3150 FOR I=1 TO 12
3160 READ T$
3170 GOSUB 3520
3180 V$(I)=TN$
3190 NEXT I
3200 FOR I=1 TO 6
3210 READ T$
3220 GOSUB 3520
3230 P$(I)=TN$
3240 NEXT I
3250 FOR I=1 TO 36
3260 READ T$
3270 GOSUB 3520
3280 R$(I)=TN$
3290 NEXT I
3300 FOR I=1 TO 12
3310 READ VP(I)
3320 NEXT I
3330 FOR I=1 TO 6
3340 READ PR(I,1):: PR(I,2)=1
3350 READ PR(I,3)
3360 NEXT I
3370 FOR I=1 TO 8
3380 READ D(I)
3390 NEXT I
3400 IN=0
3410 RETURN
3420 DATA 21,1,22,2,22,1,23,2,18,1,19,2,16,3,20,4,11,1,12,2,7,1,8,2,7,4,3,3,2,1
3430 DATA EST,QUEST,NORD,SUD,HAUT,BAS,TUER,HELP,PREND,POSE,INVENTAIRE,PANSER,ALL
UMPROJ,ETEINPROJ,LIRE
3440 DATA QSPKFDUFVS,TBD!B!GBSJOF,CJEPO,NBOVFMEFDVJTJOF,GFVJMMFU,QBOTFNFOU,HMBJW
F,HBUFV,DPSQTEFJU,DBEBSFHFBOU,TFSQFOUNPSU,CPVDNPSU
3450 DATA J/U/,NPOTUSF,TFSQFOU,CPVD,DIBVESPO,PCVT

```

```

3460 DATA USPVP!EFQBSU,QPJOU!OJE,DPVMPJS!TFDSFU,RH,USPV!P,USPV!S,USPV!J,DBCJOFU!O
PJS,USPV!V,USPV!W,USPV!S,WJEF
3470 DATA QFUJU!TBMP0,USPV!SPDIFVY,MJFV!PEPSBOU,ESBHP0!OJE,TFSQFOU!OJE,TBMP0!KBV
OF,UBMVT,MJFV!QVBOU,SFGVHF,USPV!BSSJWFF
3480 DATA USPVP!DPMPSF,USPV!HFMF,MJFV!PVWFSU,CPJT,CPJT,CPJT,CPJT,CPJT,CPJT,C
PJT,CPJT,CPJT,CPJT
3490 DATA 34,30,28,21,14,15,13,0,0,0,0
3500 DATA 34,2,16,15,17,4,29,2,8,1,25,1
3510 DATA 1,5,9,10,11,7,6,2
3520 REM
3530 TN$=""
3540 FOR I1=1 TO LEN(T$)
3550 C$=SEG$(T$,I1,1)
3560 C$=CHR$(ASC(C$)-1)
3570 TN$=TN$&C$
3580 NEXT I1
3590 IF IN=0 THEN PRINT TN$
3600 RETURN
10000 FOR I=1 TO 12 :: PRINT V$(I):: NEXT I

```

LE LOUP ET LES AGNEAUX

Un conte de La Fontaine? Non, mais un jeu étonnant. Il se joue sur un damier visualisé sur l'écran: lorsque le jeu commence, les agneaux sont disséminés sur la moitié inférieure du damier et représentés par des petites cases contenant un nombre. Le loup se tient dans le coin en haut à gauche.

Nous partons - vous êtes le loup, et l'ordinateur commande les agneaux. Si vous dévorez 3 agneaux, vous gagnez; par contre, si un des agneaux vous mange, c'est l'ordinateur qui l'emporte (ces agneaux mangeurs de loup sont la revanche de celui de la Fontaine).

Vous pouvez commencer. Sur la partie inférieure de l'écran vous voyez:

VOUS POUVEZ FAIRE ... PAS

par exemple

VOUS POUVEZ FAIRE 2 PAS

Le nombre de "pas" est toujours compris entre 1 et 3 et vous indique combien de mouvements vous pouvez faire pendant votre tour. Le loup - c'est-à-dire vous - peut se déplacer horizontalement ou verticalement, mais jamais en diagonale. Chaque mouvement doit être entré avec les touches de déplacement du curseur. Si le **dernier** déplacement vous mène sur une case occupée par un agneau, il est à vous. Vous ne devez jamais traverser une case sur laquelle se trouve un agneau.

Il y a 5 agneaux en tout; ils peuvent jouer à saute-mouton mais ne doivent se déplacer que dans une direction.

Le nombre de pas qu'ils peuvent faire, indiqué par leur numéro est toujours compris entre 1 et 5. Par exemple, si celui qui porte le numéro 3 peut faire

3 pas vers la gauche, ou
3 pas vers la droite, ou
3 pas en avant, ou
3 pas en arrière

Il ne peut pas faire 1 pas en avant et 2 vers la gauche. Si vous (le loup) avez atteint votre nouvelle position, l'ordinateur indique

MAINTENANT C'EST MON TOUR

Quand c'est votre tour, l'ordinateur affiche

VOUS POUVEZ FAIRE ... PAS

A la fin du jeu, l'ordinateur vous dit

VOUS AVEZ GAGNE!

ou pire

VOUS AVEZ PERDU!

```
10 REM TI(ETENDU)-LE LOUP ET LES AGNEAUX
20 CALL CLEAR
30 RANDOMIZE
40 S1$="VOUS POUVEZ FAIRE  PAS"
50 S2$="MAINTENANT C'EST A MOI"
60 GOTO 810
70 REM MOUVEMENT DU LOUP***
80 DISPLAY AT(20,2):S1$
90 CALL HCHAR(20,22,BW+48)
100 CALL KEY(1,KE,ST)
110 IF ST=0 OR ST=-1 THEN 100
120 IF KE=0 THEN RW=RW+1 :: KWN=KW :: GOTO 160
130 IF KE=2 THEN RW=RW :: KWN=KW-1 :: GOTO 160
140 IF KE=5 THEN RW=RW-1 :: KWN=KW :: GOTO 160
150 IF KE=3 THEN RW=RW :: KWN=KW+1 ELSE 100
160 IF RW<RMIN OR RW>RMAX OR KWN<KMIN OR KWN>KMAX THEN 100
170 CALL GCHAR(RW,KWN,GC)
180 IF GC>48 AND GC<54 AND BW<>1 THEN 100
190 BW=BW-1 :: CALL HCHAR(20,22,BW+48)
200 CALL HCHAR(RW,KW,30):: KW=KWN :: RW=RW
210 CALL HCHAR(RW,KW,ASC("W"))
220 IF BW=0 THEN 100
230 FOR I=1 TO 5
240 IF RW<>P(I,1)OR KW<>P(I,2)THEN 260
250 IF P(I,3)<>0 THEN P(I,3)=0 :: BEAT=BEAT+1
260 NEXT I
270 BW=INT(RND*3+1)
280 RETURN
```



```

290 REM MOUVEMENT AGNEAU**
300 DISPLAY AT(20,2):S2$
310 MMAX=-400
320 FOR I=1 TO 5
330 IF P(I,3)=0 THEN 420
340 KS=P(I,2):: RS=P(I,1)-P(I,3)
350 GOSUB 500
360 RS=P(I,1)+P(I,3)
370 GOSUB 500
380 RS=P(I,1):: KS=P(I,2)+P(I,3)
390 GOSUB 500
400 KS=P(I,2)-P(I,3)
410 GOSUB 500
420 NEXT I
430 CALL HCHAR(P(IMAX,1),P(IMAX,2),30)
440 P(IMAX,1)=RMAX
450 P(IMAX,2)=KMAX
460 P(IMAX,3)=INT(RND*5+1)
470 CALL HCHAR(RMAX,KMAX,ASC(STR$(P(IMAX,3))))
480 RETURN
490 REM EVALUATION*****
500 STAT=0
510 IF RS<RMIN OR RS>RMAX OR KS<KMIN OR KS>KMAX THEN RETURN
520 D=ABS(RW-RS)+ABS(KW-KS)
530 IF D=BW THEN RETURN
540 IF D=BW-2 THEN STAT=-200

```

```

550 IF D=0 THEN STAT=500 :: GOTO 630
560 IF D<=5 THEN STAT=STAT+35
570 IF RS=RW THEN STAT=STAT+40
580 IF KS=KW THEN STAT=STAT+40
590 IF ABS(P(I,1)-RW)+ABS(P(I,2)-KW)=BW THEN STAT=STAT+60
600 FOR T=1 TO 5
610 STAT=STAT-((RS<>P(T,1))+(KS<>P(T,2)))*10
620 NEXT T
630 IF STAT+RND<MMAX THEN RETURN
640 MMAX=STAT :: IMAX=I :: RMAX=RS :: KMAX=KS
650 RETURN
660 REM INITIALISATION ET TRACE DAMIER*****
670 RBMIN=7 :: RBMAX=16 :: KBMIN=11 :: KBMAX=20
680 BW=2 :: MMAX=-400 :: BEAT=0
690 RW=RBMIN :: KW=KBMIN
700 FOR I=0 TO 9
710 CALL HCHAR(RBMIN+I,KBMIN,30,10)
720 NEXT I
730 FOR I=1 TO 5
740 P(I,1)=INT(RND*5)+RBMIN+5
750 P(I,2)=INT(RND*10)+KBMIN
760 P(I,3)=INT(RND*5)+1
770 CALL HCHAR(P(I,1),P(I,2),ASC(STR$(P(I,3))))
780 NEXT I
790 CALL HCHAR(RW,KW,ASC("W"))
800 RETURN
810 REM PROGRAMME PRINCIPAL***
820 GOSUB 670
830 GOSUB 80
840 IF BEAT=3 THEN PRINT "VOUS AVEZ GAGNE" :: END
850 GOSUB 300
860 IF MMAX<400 THEN 830
870 PRINT "VOUS AVEZ PERDU"
880 END

```

AUTOROUTE

Ce jeu est une démonstration du pouvoir des lutins. Après avoir entré RUN, vous allez voir une autoroute occupée par de nombreuses voitures - votre tâche est de la traverser sain et sauf.

Utilisez le manche à balai pour contrôler vos mouvements (la touche ALPHA LOCK ne doit pas être enclenchée sinon le levier de commande ne fonctionne plus correctement).

Au début du jeu l'ordinateur vous demande un niveau de vitesse pour les voitures. Elles ne roulent pas toutes à la même vitesse, ce qui vous complique la tâche.

Si vous avez le malheur de vous retrouver impliqué dans un accident, l'ordinateur demande.

ENCORE?

et en pressant ENTER vous voilà miraculeusement remis sur pied; le jeu peut reprendre.

```
10 REM TI(ETENDU)-AUTOROUTE
20 CALL CLEAR
30 INPUT "NIVEAU ( <25 )=":L
40 CALL CLEAR
50 S=0
60 DISPLAY AT(4,3):"NIVEAU=";L
70 CALL SCREEN(16)
80 CALL COLOR(9,9,9)
90 CALL HCHAR(11,1,97,32)
100 CALL HCHAR(20,1,97,32)
110 CALL CHAR(42,"00F8A8F8FFFF6363")
120 A$="18187EBDBD3C2424"
130 CALL CHAR(90,A$)
140 FOR K=1 TO 8
150 R=INT(RND*30+10)
160 CALL SPRITE(8K,42,K+3,81+8*K,8,0,R+L)
170 NEXT K
180 CALL SPRITE(820,90,2,170,80+K*8)
190 CALL JOYST(1,X,Y)
```

```
200 CALL MOTION(#20,-3*Y,3*X)
210 CALL COINC(ALL,M)
220 IF M=-1 THEN GOTO 260
230 CALL POSITION(#20,YL,XL)
240 IF YL<85 THEN 330
250 GOTO 190
260 CALL MOTION(#20,0,0)
270 CALL SOUND(2000,-7,5)
280 DISPLAY AT(8,8):"ENCORE?"
290 ACCEPT AT(8,16):Q$
300 IF SEG$(Q$,1,1)="N" THEN 370
310 CALL DELSPRITE(ALL)
320 GOTO 20
330 S=S+1
340 DISPLAY AT(5,3):"SCORE=";S
350 CALL DELSPRITE(#20)
360 GOTO 180
370 END
```

FORMULE 1

Mesdames et Messieurs, les voitures sont maintenant sur la ligne de départ. Voici le plus passionnant Grand Prix qu'il vous ait été donné de voir. Puisque l'enjeu en est le titre de champion du monde, tous les pilotes ne viseront que la première place. Il fait beau, le circuit est sec, ce sera donc une très belle course. Qui sait, nous verrons peut-être même aujourd'hui le record du tour battu. Une fois les vérifications de dernière minute faites, les concurrents font chauffer leurs moteurs jusqu'à pleine puissance, le drapeau de départ est abaissé; ils sont partis! Après le RUN, l'ordinateur demande

1 OU 2 JOUEURS?

Puis il vous demande si vous avez un manche à balai. Dans le cas contraire, dirigez vos voitures avec

Z		U
Q	S	H J
W		N
voiture 1		voiture 2

Il vous demande enfin le nombre de tours que vous voulez faire. Vous vous rendrez vite compte que ces voitures sont soumises aux mêmes lois physiques que dans la réalité. Ceci rend le jeu assez difficile, alors n'hésitez pas à remettre cent fois sur le métier l'ouvrage.

```
10 REM TI(ETENDU)-FORMULE 1
20 CALL CLEAR
30 OPTION BASE 1
40 DIM CL(2),FT(2),DT(2),R0(2),MT(2),LT(2),XOLD(2),YOLD(2),XNEW(2),YNEW(2),VX(2)
,VY(2),ACCX(2),ACCY(2),CF(2),S0(2),RE(1)
50 LA=1 :: RA=2 :: UA=4 :: DA=15
```



```


60 FR=.9 :: CL(1)=9 :: CL(2)=5 :: F1=50/60 :: F2=60*F1 :: RE(1)=300
70 GOTO 1320
80 REM **INITIALISATION
90 CALL CHAR(96,"E047427F7F4247E0")
100 CALL CHAR(97,"07E242FEFE42E207")
110 CALL CHAR(98,"5A7E5A181899FFB1")
120 CALL CHAR(99,"81FF9918185A7E5A")
130 CALL CHAR(33,"FFFEFCF8F0E0C080")
140 CALL CHAR(34,"0103070F1F3F7FFF")
150 CALL CHAR(35,"FF7F3F1F0F070301")
160 CALL CHAR(36,"80C0E0F0F8FCFEFF")
170 CALL CHAR(37,"FFFFFFFFFFFFFFFF")
180 CALL CHAR(62,"0000000402FF0204")
190 CALL CLEAR :: CALL SCREEN(13):: CALL COLOR(1,1,1):: CALL HCHAR(1,1,37,768)
200 CALL COLOR(1,1,12,2,2,12)
210 READ X,PX,PY :: FOR I=1 TO X
220 READ DIR,LN :: IF DIR=0 THEN 240
230 ON DIR GOTO 250,260,270,280,290,300,310,320
240 PX=PX+LN :: READ EXTRA :: PY=PY+EXTRA :: GOTO 220
250 CALL HORIZONTAL(PX,PY,LN):: PX=PX+LN :: GOTO 360
260 PX=PX+LN :: PY=PY-LN :: CALL DIAGONAL(PX-1,PY,-1,LN):: GOTO 360
270 PY=PY-LN :: CALL VERTICAL(PX,PY,LN):: GOTO 360
280 PX=PX-LN :: PY=PY-LN :: CALL DIAGONAL(PX,PY,1,LN):: GOTO 360
290 PX=PX-LN :: CALL HORIZONTAL(PX,PY,LN):: GOTO 360
300 CALL DIAGONAL(PX-1,PY,-1,LN):: PX=PX-LN :: PY=PY+LN :: GOTO 360
310 CALL VERTICAL(PX,PY,LN):: PY=PY+LN-3 :: GOTO 360
320 CALL DIAGONAL(PX,PY,1,LN):: PX=PX+LN :: PY=PY+LN :: GOTO 360
330 DATA 19,7,1
340 DATA 1,8,8,3,1,1,2,3,1,2,8,7,0,-3,0,7,12,0,3,0,6,5,5,11
350 DATA 3,10,0,3,0,4,3,5,3,6,3,7,9,0,3,0,6,4,5,1,4,4,0,0,3,3,16,2,4
360 NEXT I
370 CALL CHAR(62,"0000000402FF0204"):: CALL HCHAR(4,8,62)
380 CALL CHAR(45,"0030300000303000"):: CALL VCHAR(1,8,45,3)
390 RETURN
400 REM **START
410 DISPLAY ERASE ALL AT(3,3):"1 OU 2 JOUEURS ? 2"
420 ACCEPT AT(3,24)SIZE(-1)BEEP VALIDATE("12"):NP
430 DISPLAY AT(5,3):"MANCHE A BALAI? 0" :: ACCEPT AT(5,20)BEEP SIZE(-1)VALIDATE
("ONon"):AN$
440 JO=(AN$="0")OR(AN$="o")
450 DISPLAY AT(7,3):"NOMBRE DE TOURS? 1" :: ACCEPT AT(7,20)BEEP SIZE(-1)VALIDATE
(DIGIT):NR
460 NR=INT(NR):: IF NR<1 THEN 450
470 FOR PL=1 TO 2

```

```

480 XNEW(PL)=54 :: VX(PL)=0 :: XOLD(PL)=54
490 YNEW(PL)=12*PL-4 :: VY(PL)=0 :: YOLD(PL)=YNEW(PL)
500 DT(PL)=0 :: MT(PL)=1E6 :: LT(PL)=0 :: RO(PL)=0
510 SO(PL)=40000 :: CF(PL)=1 :: NEXT PL
520 NF=0 :: MI=0
530 DC=1 :: RETURN
540 FOR PL=1 TO NP
550 CALL SPRITE(#PL,96,CL(PL),YNEW(PL)-4,XNEW(PL)-4)
560 NEXT PL
570 RETURN
580 REM **ENTREE ACCELERATION***
590 IF JO THEN 660
600 ACCX(PL)=0 :: ACCY(PL)=0 :: CALL KEY(PL,K,X)
610 IF K=LA THEN ACCX(PL)=-4
620 IF K=RA THEN ACCX(PL)=4
630 IF K=UA THEN ACCY(PL)=4
640 IF K=DA THEN ACCY(PL)=-4
650 RETURN
660 CALL JOYST(PL,ACCX(PL),ACCY(PL))
670 RETURN
680 REM **CALCUL NOUVELLES VITESSE ET PLACE
690 VX(PL)=(VX(PL)+ACCX(PL))*FR
700 VY(PL)=(VY(PL)-ACCY(PL))*FR
710 XOLD(PL)=XNEW(PL):: XNEW(PL)=XNEW(PL)+VX(PL)
720 YOLD(PL)=YNEW(PL):: YNEW(PL)=YNEW(PL)+VY(PL)
730 RETURN
740 REM **MOUVEMENT
750 XINT=INT(XNEW(PL)):: YINT=INT(YNEW(PL)):: IF XINT<5 OR YINT<5 OR XINT>255 OR
YINT>188 THEN 820
760 CALL GCHAR(INT(YINT/8+1),INT(XINT/8+1),CH):: IF CH=32 OR CH=45 THEN 850
770 IF CH>36 THEN 820
780 CALL CHARPAT(CH,CHAR$)
790 CTL$=SEG$(CHAR$,INT((YINT/8-INT(YINT/8))*16+1),2)
800 CALL HEXBIN(CTL$,BIN$)
810 IF SEG$(BIN$,(XINT/8-INT(XINT/8))*8+1,1)="0" THEN 850
820 XNEW(PL)=XOLD(PL):: YNEW(PL)=YOLD(PL)
830 DT(PL)=INT(ABS(VX(PL))+ABS(VY(PL)))
840 CALL MOTION(#PL,0,0):: VX(PL)=0 :: VY(PL)=0 :: RETURN
850 IF ABS(VY(PL))>ABS(VX(PL)) THEN 890
860 IF VX(PL)<0 THEN CVAL=97 :: GOTO 910
870 IF VX(PL)>0 THEN CVAL=96 :: GOTO 910
880 GOTO 920
890 IF VY(PL)<0 THEN CVAL=98 :: GOTO 910
900 CVAL=99

```



```

910 CALL PATTERN(#PL,CVAL)
920 CALL LOCATE(#PL,YNEW(PL)-4,XNEW(PL)-4)
930 S0(PL)=10*(ABS(VX(PL))+ABS(VY(PL)))+109 :: IF S0(PL)<110 THEN S0(PL)=40000
940 CALL MOTION(#PL,VY(PL)/5,VX(PL)/5)
950 RETURN
960 REM **TERMINE?
970 TIMER=TIMER+1 :: TI=TIMER/F1+MI
980 IF TI-MI>F2 THEN TIMER=TIMER-F2 :: MI=MI+60
990 FOR PL=1 TO 2
1000 ON CF(PL)GOTO 1010,1030,1050,1070
1010 IF XNEW(PL)>200 THEN CF(PL)=2 :: GOTO 1130
1020 GOTO 1130
1030 IF YNEW(PL)>150 AND XNEW(PL)<150 THEN CF(PL)=3 :: GOTO 1130
1040 GOTO 1130
1050 IF XNEW(PL)<50 THEN CF(PL)=4 :: GOTO 1130
1060 GOTO 1130
1070 IF NOT(XNEW(PL)>50 AND YNEW(PL)<30) THEN 1130
1080 CF(PL)=1 :: RT=TI-LT(PL):: LT(PL)=TI
1090 IF RT<MT(PL) THEN MT(PL)=RT
1100 RO(PL)=RO(PL)+1 :: IF RO(PL)<NR THEN 1130
1110 NF=NF+1 :: FT(PL)=TI
1120 FOR S1=1 TO 3 :: FOR S2=200 TO 700 STEP 50 :: CALL SOUND(-99,S2,0):: NEXT S
2 :: NEXT S1
1130 NEXT PL
1140 RETURN

```

```

1150 REM **FIN
1160 CALL CLEAR :: CALL CHARSET :: CALL DELSPRITE(ALL):: CALL SCREEN(11)
1170 PRINT "TEMPS RECORD:";:: TI=RE(DC):: GOSUB 1290
1180 PRINT :: PRINT "NOMBRE DE TOURS: ";NR :: PRINT
1190 FOR PL=1 TO NP
1200 PRINT :: PRINT
1210 PRINT "JOUEUR ";PL :: PRINT
1220 PRINT "TOTAL: ";:: TI=FT(PL):: GOSUB 1290
1230 PRINT "MOYENNE:";:: TI=FT(PL)/RO(PL):: GOSUB 1290
1240 PRINT "RECORD :";:: TI=MT(PL):: GOSUB 1290
1250 IF TI<RE(DC)THEN RE(DC)=TI :: PRINT "NOUVEAU RECORD DU TOUR!!!" :: PRINT
1260 NEXT PL
1270 RETURN
1280 REM **TEMPS FINAL**
1290 MI=INT(TI/60):: PRINT USING " ##:##.##":MI,TI-60*MI
1300 RETURN
1310 REM **PROGRAMME PRINCIPAL**
1320 GOSUB 410
1330 RESTORE :: GOSUB 90 :: GOSUB 540
1340 FOR PL=1 TO 2
1350 IF PL>NP THEN CALL SOUND(-50,S0(1),0):: CALL SOUND(1000,S0(1),1):: GOTO 1400
1360 IF DT(PL)>0 THEN DT(PL)=DT(PL)-1 :: S0(PL)=-6 :: GOTO 1400
1370 GOSUB 590
1380 GOSUB 690
1390 GOSUB 750
1400 NEXT PL
1410 IF S0(1)<>S0(2)THEN CALL SOUND(-4000,S0(1),1,S0(2),1):: GOTO 1430
1420 CALL SOUND(-4000,S0(1),1)
1430 GOSUB 970
1440 IF NF<NP THEN 1340
1450 GOSUB 1160
1460 PRINT :: INPUT "ON REJOU (O/N)? ":AN$
1470 AN$=SEG$(AN$,1,1):: IF AN$="O" OR AN$="o" THEN 1320
1480 END
1490 SUB HORIZONTAL(STX,STY,LENGHT)
1500 FOR I=0 TO 2
1510 CALL HCHAR(STY+I,STX,32,LENGHT)
1520 NEXT I
1530 SUBEND
1540 SUB VERTICAL(STX,STY,LENGHT)
1550 FOR I=0 TO 2
1560 CALL VCHAR(STY,STX+I,32,LENGHT)
1570 NEXT I

```

```

1580 SUBEND
1590 SUB DIAGONAL(STX,STY,DX,LENGHT)
1600 IF DX=1 THEN CH1=35 :: CH2=36 :: GOTO 1630
1610 IF DX=-1 THEN CH1=33 :: CH2=34 :: GOTO 1630
1620 SUBEXIT
1630 FOR I=0 TO LENGHT-1
1640 X=STX+DX*I :: Y=STY+I
1650 CALL GCHAR(Y,X,CH):: IF CH=37 THEN CALL HCHAR(Y,X,CH1)
1660 CALL VCHAR(Y+1,X,32,2)
1670 CALL GCHAR(Y+3,X,CH):: IF CH=37 THEN CALL HCHAR(Y+3,X,CH2)
1680 NEXT I
1690 SUBEND
1700 SUB HEXBIN(HEX$,BIN$)
1710 BIN$=""
1720 FOR I=LEN(HEX$)TO 1 STEP -1
1730 A$=SEG$(HEX$,I,1):: IF ASC(A$)>64 THEN 1750
1740 DEC=VAL(A$):: GOTO 1760
1750 DEC=ASC(A$)-55
1760 FOR J=1 TO 4
1770 N=INT(DEC/2):: BIN$=STR$(INT(DEC-2*N+.1))&BIN$
1780 DEC=N :: NEXT J
1790 NEXT I
1800 SUBEND

```

TIROIR-CAISSE

Avez-vous remarqué, en faisant vos courses, comme les commerçants allaient vite pour vous faire l'addition. Jouez à ce jeu avec vos amis pour voir si vous pourriez faire un bon marchand.

Vous allez découvrir des paires de nombres de longueur croissante qu'il vous faut additionner. Vous allez vite vous rendre compte que ce n'est pas aussi simple qu'il y paraît. Pour comprendre pourquoi, examinez l'addition suivante:

$$\begin{array}{r} 75856 \\ + 37637 \\ \hline \end{array}$$

Normalement vous additionnez d'abord les nombres de la colonne de droite, puis ceux de la colonne suivante et ainsi de suite. Avec l'ordinateur, en revanche, vous devez entrer la réponse en commençant par la colonne de gauche. C'est cette différence qui fait tout l'intérêt du jeu.

Combien de nombres pouvez-vous additionner correctement en une minute? Notre record est de 8, tâchez de le battre!

```
10 REM TI-TIROIR-CAISSE
20 RANDOMIZE
30 DIM J0$(9)
40 FOR I=0 TO 9
50 READ J0$(I)
60 NEXT I
70 CALL CLEAR
80 NR=0
90 TI=0
100 MX=1
110 GOTO 600
120 DATA BOF,PAS TERRIBLE,PAS TROP MAL,PAS MAL,BIEN,TRES BIEN,REMARQUABLE,FANTAS
TIQUE,CHAMPION,DU JAMAIS VU
140 REM**ADDITION AU HASARD**
150 N1$=STR$(INT(RND*MX))
160 N2$=STR$(INT(RND*MX))
```



```

170 AN=VAL(N1$)+VAL(N2$)
180 L1=LEN(N1$)
190 L2=LEN(N2$)
200 D=L1-L2
210 FOR I=1 TO ABS(D)
220 IF L1=L2 THEN 270
230 IF L2<L1 THEN 260
240 N1$=" "&N1$
250 GOTO 270
260 N2$=" "&N2$
270 NEXT I
280 PRINT N1$
290 PRINT N2$;"+"
300 FOR DU=0 TO NR+1
310 PRINT "-";
320 NEXT DU
330 RETURN
340 REM**ENTREE REPONSE**
350 AN$=""
360 PRINT "REPONSE:";
370 TI=TI+1
380 IF TI=1500 THEN 530
390 CALL KEY(3,DI,S)
400 IF S=0 THEN 370
410 IF DI<>8 THEN 450
420 IF AN$="" THEN 370
430 PRINT "/"&SEG$(AN$,LEN(AN$),LEN(AN$))&"/";
440 GOTO 490
450 PRINT CHR$(DI);

```

```


460 IF DI<48 THEN 490
470 IF DI>57 THEN 370
480 AN$=AN$&CHR$(DI)
490 IF DI<>8 THEN 510
500 AN$=SEG$(AN$,1,LEN(AN$)-1)
510 IF DI=13 THEN 540
520 GOTO 370
530 PRINT
540 RETURN
550 REM**CONTROLE REPONSE**
560 IF AN$<>" " THEN 580
570 AN$="-1"
580 IF VAL(AN$)=AN THEN 700 ELSE 740
590 RETURN
600 REM**PROGRAMME PRINCIPAL**
610 MX=MX*10
620 GOSUB 140
630 GOSUB 340
640 GOSUB 550
650 IF VAL(AN$)<>AN THEN 670
660 IF NR<10 THEN 610
670 PRINT "RESULTAT: "
680 PRINT JO$(NR)
690 END
700 REM**BONNE REPONSE**
710 PRINT "BRAVO!"
720 NR=NR+1
730 GOTO 590
740 REM**REPONSE FAUSSE**
750 PRINT "FAUX !"
760 PRINT "C'EST: ";AN
770 GOTO 590

```


LA TRAVERSEE

Simple en apparence, ce jeu demande de la concentration. Vous allez voir 8 barres horizontales avec des trous. Au-dessus des barres il y a 8 pions qui peuvent tomber à travers ces trous.

Dans l'exemple suivant



le pion numéro 2 est déjà tombé. Le but du jeu est de faire passer les 8 pions par les trous. Pour obtenir ce résultat, il faut déplacer les barres de façon à ce que les trous soient sous les pions.

Pour faire bouger une barre, entrez un ordre ayant la forme suivante

BDE

dans laquelle

B est le numéro de la barre: les barres sont numérotées de 1 (en haut) à 8 (en bas)

D est la direction dans laquelle vous voulez la déplacer (G à gauche, D à droite) et

E le nombre de déplacements élémentaires que la barre doit faire.

Par exemple

3D4

fera bouger la barre n° 3 de 4 pas vers la droite. Tout cela semble bien simple mais essayez et vous vous apercevrez que ce jeu est un bon test de vos capacités de raisonnement logique.

```


10 REM TI(ETENDU)-TRAVERSEE
20 DIM MAT(9,18),CHT(9,18)
30 GOSUB 550
40 ROW=1
50 GOSUB 300
60 GOSUB 430
70 REM PROGRAMME PRINCIPAL*****
80 GOSUB 690
90 FOR CNT=1 TO AZ
100 IF RI=-1 THEN 200
110 FOR CNT2=17 TO 1 STEP -1
120 CHT(ROW,CNT2+1)=CHT(ROW,CNT2)
130 MAT(ROW,CNT2+1)=MAT(ROW,CNT2)
140 NEXT CNT2
150 CHT(ROW,1)=CHT(ROW,18)
160 MAT(ROW,1)=MAT(ROW,18)
170 GOSUB 300
180 GOSUB 430
190 GOTO 280
200 FOR CNT2=1 TO 17
210 CHT(ROW,CNT2-1)=CHT(ROW,CNT2)
220 MAT(ROW,CNT2-1)=MAT(ROW,CNT2)
230 NEXT CNT2
240 CHT(ROW,17)=CHT(ROW,0)
250 MAT(ROW,17)=MAT(ROW,0)
260 GOSUB 300
270 GOSUB 430
280 NEXT CNT
290 GOTO 80
300 REM PLACEMENT PIONS****
310 FOR C1=1 TO 17
320 IF MAT(ROW,C1)=-1 THEN GOSUB 360
330 NEXT C1
340 RETURN
350 REM LES PIONS TOMBENT**
360 OG=9
370 FOR C2=8 TO 1 STEP -1
380 IF MAT(C2,C1)=-1 AND CHT(C2,C1)=0 THEN 400
390 OG=C2-1 :: GOTO 410
400 IF CHT(C2-1,C1)>0 THEN CHT(OG,C1)=CHT(C2-1,C1):: CHT(C2-1,C1)=0 :: OG=OG-(OG
=9)-1
410 NEXT C2
420 RETURN

```

```

430 REM DESSIN ECRAN*****
440 NT=NT+1
450 FOR C6=0 TO 8
460 FOR C7=1 TO 17
470 IF MAT(C6,C7)=0 THEN CALL HCHAR(5+C6,5+C7,128):: GOTO 500
480 IF CHT(C6,C7)<>0 THEN CALL HCHAR(5+C6,5+C7,CHT(C6,C7)+48):: GOTO 500
490 CALL HCHAR(5+C6,5+C7,31)
500 NEXT C7
510 CALL HCHAR(5+C6,3,C6+48)
520 NEXT C6
530 DISPLAY AT(23,18)BEEP:NT
540 RETURN

```



```

550 REM INITIALISATION*****
560 NT=-1
570 FOR CC=1 TO 9
580 CHT(0,2*CC-1)=CC :: MAT(0,2*CC-1)=-1 :: MAT(0,CC*2)=1
590 NEXT CC
600 CALL CHAR(128,"00FFFFFFFFFFFFFF")
610 CALL CLEAR
620 PRINT "NOMBRE DE COUPS: "
630 FOR TT=1 TO 8
640 FOR TTT=1 TO 4
650 MAT(TT,INT(RND*17+1))=-1
660 NEXT TTT
670 NEXT TT
680 RETURN

```

```

690 REM ENTREE MOUVEMENT***
700 CALL HCHAR(18,4,63,3)
710 CALL SOUND(10,880,0)
720 FOR C3=1 TO 3
730 CALL KEY(3,K,S):: IF S=0 OR S=-1 THEN 730
740 A(C3)=K
750 CALL HCHAR(18,3+C3,K)
760 NEXT C3
770 ROW=A(1)-48 :: IF ROW<1 OR ROW>8 THEN 700
780 RI=(A(2)=71)-(A(2)=68)
790 IF RI=0 THEN 700
800 AZ=A(3)-48 :: IF AZ<1 OR AZ>9 THEN 700
810 RETURN
820 GOTO 700

```

ABORDAGE EXTRA-TERRESTRE

D'abord un mot pour décrire ce jeu le mieux possible. Dans la moitié inférieure de l'écran se trouve la forme suivante



Des petits carrés tombent du haut de l'écran. Vous devez faire bouger la forme en utilisant les touches de contrôle



pour empêcher les carrés de la heurter. Cela paraît trivial, mais attendez la suite ...


Vous êtes le capitaine d'un des plus beaux astronefs de l'univers. En traversant majestueusement sur la voie lactée, vous ne pouvez vous empêcher d'être fier de ce magnifique vaisseau que vous commandez. Puis, alors que vous approchez du port et de la fin de votre voyage, c'est le désastre:

UNE ATTAQUE DE SOUCOUPES VOLANTES

Comment échapper à ces envahisseurs ennemis venus de l'autre bout de l'univers? Peut-être en essayant votre

FUYARD SUPER ATOMIQUE

Vous ferez autant de points que vous éviterez de soucoupes volantes. Réconfortant, n'est-ce pas? En tout cas, ce jeu est rapide, simple et très amusant, et c'est ce qui compte.



```
10 REM TI(ETENDU)-ABORDAGE EXTRA-TERRESTRE
20 DIM R(16)
30 CALL CLEAR
40 INPUT "NIVEAU(1-5)?":LE
50 CALL MAGNIFY(2)
60 SC=0
70 CALL COLOR(3,16,1)
80 CALL COLOR(4,16,1)
90 A$="003C7EDB7E3C0000"
100 B$="0000000081423C00"
110 CALL CHAR(42,A$)
```

```

120 CALL CHAR(95,B$)
130 CALL CLEAR
140 CALL SCREEN(2)
150 CALL SPRITE(#20,95,16,150,75)
160 FOR K=3 TO 16
170 R(K)=INT(RND*10+2*LE)
180 CALL SPRITE(#K,42,K,16,17*K-41,R(K),0)
190 GOSUB 410
200 NEXT K
210 FOR K=3 TO 16
220 GOSUB 410
230 IF M=-1 THEN K=16
240 SC=SC+1
250 R(K)=R(K)+LE
260 CALL MOTION(#K,R(K),0)
270 DISPLAY AT(3,3):SC
280 CALL SOUND(5,440,5)
290 NEXT K
300 IF M<>-1 THEN GOTO 210
310 CALL SOUND(2000,-7,5)
320 CALL SCREEN(12)
330 CALL DELSPRITE(ALL)
340 CALL COLOR(3,2,1)
350 CALL COLOR(4,2,1)
360 DISPLAY AT(20,2):"ON REJOU(0/N)?"
370 ACCEPT AT(20,18):Q$
380 IF SEG$(Q$,1,1)="0" OR SEG$(Q$,1,1)="o" THEN GOTO 30
390 END
400 REM MOUVEMENT VAISSEAU***
410 CALL JOYST(1,X,Y)
420 CALL MOTION(#20,0,3*X)
430 CALL COINC(ALL,M)
440 RETURN


```

MINI-MANCALA

MINI MANCALA, création de C. Freelink, est basé sur un jeu arabe très ancien. On y joue en faisant passer des pierres d'une coupe à une autre. Il y a 4 coupes en tout. A et B sont celles de l'ordinateur, C et D les vôtres. Au début du jeu, il y a 2 pierres dans chaque coupe.




A (ordinateur)



B (ordinateur)




C (vous)



D (vous)

Chacun à son tour, les joueurs prennent les pierres d'une de leurs propres coupes et les distribuent aux trois autres coupes dans le sens inverse des aiguilles d'une montre. Par exemple, vous pouvez décider de déplacer les pierres de la coupe D comme ceci




A (ordinateur)



B (ordinateur)



C (vous)



D (vous)




Après cette phase la coupe D est vide. En fait, il y aura toujours au moins une coupe de vide parce qu'on n'a pas le droit de remettre les pierres dans la coupe où on les a prises.

Vous avez gagné lorsque toutes les pierres sont dans vos coupes.

Sur l'ordinateur les coupes sont visualisées par des cases sur lesquelles sont inscrits des chiffres représentant le nombre de pierres que chaque coupe contient. Vous pouvez choisir entre les niveaux de difficulté 1, 2 ou 3 (1 est le plus facile). Vous pouvez aussi décider d'être le premier à jouer. L'ordinateur demande dans quelle coupe vous voulez prélever les pierres et vous indique ses propres déplacements de pierres. La position des pierres sur l'écran est donnée après chaque coup.

Ne soyez pas surpris de voir ce jeu compliqué réalisé par un programme aussi court. Le diagramme suivant montre tous les mouvements possibles et s'avère un excellent exemple de la représentation sous forme de schéma d'un jeu stratégique.

```
10 REM TI(ETENDU)-MINI-MANCALA
20 DIM M(1,1),L(3,1)
30 GOTO 480
40 REM INITIALISATION*****
50 FOR I=0 TO 1
60 M(I,1)=2
70 M(I,0)=2
80 FOR J=0 TO 3
90 READ L(J,I)
100 NEXT J
110 NEXT I
120 DATA 0,1,1,0,0,0,1,1
130 RETURN
140 REM ECRAN*****
150 PRINT
160 PRINT "A    B"
170 PRINT M(0,0);M(0,1)
180 PRINT
190 PRINT M(1,0);M(1,1)
200 PRINT "C    D"
210 RETURN
220 INPUT "ON PREND LES PIERRES DE      C OU D ?":X$
230 IF (X$="D" OR X$="d")AND M(1,1)>0 THEN LB=2 :: RETURN
240 IF (X$="C" OR X$="c")AND M(1,0)>0 THEN LB=1 :: RETURN
250 PRINT "ENTREE ERRONEE"
```



```

260 GOTO 220
270 RETURN
280 REM A L'ORDINATEUR*****
290 PRINT "C'EST A MOI"
300 Q=M(0,0)*1000+M(0,1)*100+M(1,0)*10+M(1,1)
310 LB=3
320 IF M(0,0)=0 THEN 350
330 IF M(0,1)=0 THEN LB=0 :: GOTO 350
340 IF M(0,0)>0 AND(LEV<3 AND RND*LEV<.4)OR M(0,1)=6 OR Q=1430 OR Q=1340 OR Q=61
10 OR Q=1160 THEN LB=0
350 PRINT "JE PRENDS LES PIERRES DE ";CHR$(65-(LB=3))
360 RETURN
370 REM DIFFUSION*****
380 G=L(LB,0)
390 H=L(LB,1)

```


```

400 IF M(G,H)<>0 THEN 420
410 RETURN
420 LB=LB+1
430 IF LB>3 THEN LB=LB-4
440 I=L(LB,0):: J=L(LB,1)
450 M(G,H)=M(G,H)-1 :: M(I,J)=M(I,J)+1
460 GOTO 400
470 REM PROGRAMME PRINCIPAL*****
480 GOSUB 50
490 GOSUB 150
500 INPUT "NIVEAU(1-3) ":LEV
510 IF LEV<1 OR LEV>3 THEN 500
520 INPUT "QUI COMMENCE?          VOUS(V) OU MOI(M):":X$
530 IF SEG$(X$,1,1)="V" OR SEG$(X$,1,1)="v" THEN 570
540 GOSUB 220
550 GOSUB 380
560 GOSUB 150
570 IF M(1,1)=8 THEN PRINT "VOUS AVEZ GAGNE" :: GOTO 630
580 GOSUB 290
590 GOSUB 380
600 GOSUB 150
610 IF M(0,0)=8 THEN PRINT "J'AI GAGNE" :: GOTO 630
620 GOTO 540
630 INPUT "ON REJOU? (O/N):":X$
640 IF SEG$(X$,1,1)="O" OR SEG$(X$,1,1)="o" THEN RESTORE :: GOTO 480
650 END

```

ARRÊTEZ-LA!

Bien qu'il soit très court, ce programme produit un jeu rapide et excitant. L'écran ressemble à ceci



Dans la partie gauche se trouve une lettre. Lorsque celle-ci traverse l'écran vous devez essayer de l'arrêter dans la zone où se trouve la même lettre en tapant la touche S.

Au début du jeu l'ordinateur vous demande combien de fois vous voulez jouer. A la fin de la partie, il affiche votre score, par exemple

NOMBRE DE COUPS REUSSIS: 0

Peu importe, ça ira mieux la prochaine fois!

```
10 REM TI-ARRETEZ-LA
20 CALL CLEAR
30 INPUT "NOMBRE DE COUPS?":B
40 CALL CLEAR
50 REM **ECRAN*****
60 FOR A=1 TO 5
70 CALL VCHAR(1,19+2*A,30,24)
```

```

80 CALL HCHAR(1,20+2*A,96+A)
90 NEXT A
100 REM **DEPART DU JEU****
110 I=INT(RND*5)
120 D=INT((RND*22)/2)*2+2
130 FOR A=4 TO 32 STEP 2
140 CALL HCHAR(D,A-2,31)
150 CALL KEY(1,KEY,STA)
160 IF KEY<>2 THEN 230
170 IF 24+2*I=A THEN 200
180 CALL SOUND(100,110,0)
190 GOTO 260
200 CALL SOUND(100,2000,0)
210 G=G+1
220 GOTO 260
230 CALL HCHAR(D,A,I+65)
240 NEXT A
250 REM **TOUR SUIVANT****
260 T=T+1
270 IF B=T THEN 290
280 GOTO 110
290 PRINT "NOMBRE DE COUPS:";B
300 PRINT "NOMBRE DE COUPS REUSSIS:";G

```

B.A. BA = ABBA

Un "thema con variatione" est un air qui bien que trouvant son origine dans un air connu, a sa propre tonalité. Ce jeu est une sorte de "thema con variatione" dans la mesure où il ressemble à un autre jeu du présent ouvrage.

Regardez les 16 cases remplies avec des A et des B disposés au hasard.

B	A	B	A
A	A	A	B
A	B	A	B
B	B	A	B


Quand vous désignez une des cases, en utilisant la touche indiquée sur l'écran, toutes les lettres sur les lignes horizontales et verticales traversant cette case changent: les A se transforment en B et les B en A. Le but du jeu est de terminer avec un écran présentant l'aspect suivant:

A	B	B	A
A	B	B	A
A	B	B	A
A	B	B	A

```

10 REM TI-POP SUEDOISE
20 CALL CLEAR
30 DIM A(16)
40 REM INITIALISATION*****
50 FOR K=1 TO 16
60 READ A(K)
70 NEXT K
80 RANDOMIZE
90 FOR K1=1 TO 9
100 R=INT(RND*4+1)
110 C=INT(RND*4+1)
120 GOSUB 230
130 NEXT K1
140 REM PROGRAMME PRINCIPAL*****
150 GOSUB 430
160 INPUT "COLONNE=":C
170 IF (C<0)+(C>4)<>0 THEN 160
180 INPUT "LIGNE=":R
190 IF (R<0)+(R>4)<>0 THEN 180
200 GOSUB 230
210 GOSUB 430
220 GOTO 160
230 REM DETERMINATION*****
240 FOR H=1 TO 4
250 B(H)=(R-1)*4+H
260 B(H+4)=C+(H-1)*4
270 NEXT H
280 FOR K=1 TO 4
290 FOR J=K+1 TO 8

```



```

300 IF B(J) <> B(K) THEN 340
310 W=B(K)
320 B(K)=B(B)
330 B(B)=W
340 NEXT J
350 NEXT K
360 FOR J=1 TO 7
370 H=B(J)
380 A(H)=A(H)+1
390 IF A(H) <= 1 THEN 410
400 A(H)=0
410 NEXT J
420 RETURN
430 REM AFFICHAGE*****
440 PRINT
450 PRINT "1234"
460 FOR K=1 TO 4
470 FOR J=1 TO 4
480 X$="A"
490 IF A(J+(K-1)*4)=0 THEN 510
500 X$="B"
510 PRINT X$;
520 NEXT J
530 PRINT K
540 NEXT K
550 PRINT
560 RETURN
570 DATA 0,1,1,0,0,1,1,0,0,1,1,0,0,1,1,0

```


VOYELLES ET CONSONNES

Tous vos amis peuvent profiter de ce jeu de compétition. A la différence de la plupart des autres jeux informatiques, il n'implique pas de chiffres ou d'arithmétique. L'ordinateur vous donne 7 lettres. Lorsqu'il vous demande

voyelle ou consonne (V/C)

tapez V pour voyelle et C pour consonne. Lorsque vous avez vos 7 lettres, l'ordinateur vous met au défi de faire avec elles le plus long mot possible, en utilisant chaque lettre une seule fois. Votre temps est compté; vous le verrez s'égrener sur une horloge qui déroule l'alphabet à l'envers. Une fois qu'on a goûté à ce jeu tout simple, on ne peut plus s'en passer. Essayez et vous verrez!

```
10 REM TI-VOYELLES ET CONSONNES
20 RANDOMIZE
30 A1$="AEIOUY"
40 A2$="BCDFGHJKLMNPQRSTVWXZ"
50 CALL CLEAR
60 PRINT "VOYELLE OU CONSONNE(V/C): "
70 FOR CC=1 TO 8
80 CALL KEY(3,AW,S)
90 IF S=0 THEN 80
100 IF AW=86 THEN 130
110 IF AW=67 THEN 150
120 GOTO 80
130 C$=SEG$(A1$,INT(RND*6+1),1)
140 GOTO 160
150 C$=SEG$(A2$,INT(RND*20+1),1)
160 Z$=Z$&C$
170 CALL HCHAR(10,CC+2,ASC(C$))
180 NEXT CC
190 CALL CLEAR
200 PRINT "VOUS AVEZ LES LETTRES:"
210 PRINT Z$
220 PRINT "TAPEZ VOTRE MOT"
```

```

230 FOR T=2600 TO 1 STEP -5
240 CALL HCHAR(10,10,INT(T/100)+97)
250 CALL KEY(3,K,S)
260 IF S=0 THEN 410
270 IF K<>8 THEN 340
280 IF B=0 THEN 410
290 Z%=Z%&SEG$(W%,B,1)
300 CALL HCHAR(3,3+B,31)
310 B=B-1
320 W%=SEG$(W%,1,B)
330 GOTO 410
340 C%=CHR$(K)
350 P0=POS(Z%,C%,1)
360 IF P0=0 THEN 410
370 Z%=SEG$(Z%,1,P0-1)&SEG$(Z%,P0+1,B)
380 W%=W%&C%
390 B=B+1
400 CALL HCHAR(3,3+B,K)
410 NEXT T
420 PRINT "VOTRE SCORE:";B
430 END

```

ASTROLOGIE

Ce programme est basé sur une étude faite par un physicien et astrologue hollandais, le Dr. Ir. J. Van Slooten. Il travaillait dans les laboratoires de recherche de Philips et passait tous ses loisirs à faire de l'astrologie. Il a bâti une théorie selon laquelle la phase de la lune à l'époque de la naissance a une influence astrologique très importante sur le caractère d'une personne. Après avoir étudié les vies de centaines de gens, il en vint à conclure que

"la phase de la lune permet de prévoir jusqu'où va se développer la vie spirituelle et émotionnelle d'une personne, en particulier par rapport à son environnement social, et peut même aider à prédire le rôle qu'il ou elle jouera dans notre société."

Le diagramme montre le cycle lunaire. Comme il y a 3 phases: croissante, pleine et décroissante, on distingue 3 types de personnes.

LES INDIVIDUALISTES:

nés lors de la phase croissante, ils ont de l'énergie et aiment travailler seuls, sont très volontaires et nullement découragés par l'inconfort physique.

LES SOCIALISTES:

nés pendant la pleine lune, ils aiment vivre en communauté. Les mots-clés qui les définissent le mieux sont comparer, choisir et combiner. La coopération est un facteur dominant dans leur vie mais ils connaissent aussi rivalité et envie.

LES CONSERVATEURS:

nés pendant la phase décroissante, ils savent que la lumière s'en va, mais aussi qu'elle revient. Ils sont prudents, sûrs d'eux et aiment tout contrôler.

Ces trois types peuvent encore être subdivisés en 12 catégories.

Les pionniers

sont des chercheurs, toujours à l'avant-garde. La conviction qu'ils mettent à atteindre leur but peut les faire paraître bornés.



Les coordinateurs

aiment la compagnie des autres, mais se sentent "plus égaux" que leurs compagnons. Ils sont romantiques et très attachés à la vie de famille.



Les réalistes

ont une perception du monde très terre à terre! Ils se plaisent à penser que tout peut s'expliquer en termes purement physiques.



Les spiritualistes

ont l'air de venir d'une autre planète. Philosophes, ils aiment prospecter l'inconnu et l'étrange.



Les artistes

transforment la matière de façon créative. Ils créent des objets d'art, de nouveaux produits et d'une façon générale, sont des innovateurs.



Les apôtres

ne sont pas des créateurs eux-mêmes, mais diffusent les idées d'autrui dans le monde.



Les secouristes


assument des responsabilités que les autres ignorent. Ils sont attentifs aux autres.



Les chefs


prennent aussi des responsabilités mais c'est pour diriger la vie et les actions d'autrui. Ils vont de l'avant, là où la majorité préfère se tenir prudemment sur la réserve.





Les chevaliers

sont des aventuriers intrépides toujours à la recherche de nouveautés à explorer. Idéalistes, ils ont un solide sens de l'honneur.



Les arbitres

aiment superviser les actions des autres. Ils ne préjugent pas des résultats, mais lorsqu'ils ont pris une décision, ils entendent la faire respecter.



Les révolutionnaires

ne se satisfont pas du monde tel qu'il est. Ils ont tendance à voir le mauvais côté des choses et veulent transformer le monde.



Les prophètes

contemplant le monde d'en haut, apportant la sagesse du passé. Ils ne sont, bien sûr, pas toujours écoutés.



Ce programme vous dira à quelle phase de la lune vous êtes nés. En la découvrant, ce sont les secrets de votre caractère qui vous seront révélés!

```
10 REM TI-ASTROLOGIE
20 GOTO 290
30 REM ENTREE DATE*****
40 CALL CLEAR
50 PRINT "PHASE LUNAIRE"
60 PRINT
70 INPUT "JOUR      JJ: ":DD
80 INPUT "MOIS      MM: ":MM
90 INPUT "ANNEE AAAA: ":YYYY
100 M=M-12*(M<3)
110 Y=Y+(M>12)
120 RETURN
130 REM CALCUL*****
140 T=INT(365.25*YYYY)+INT(30.6*(MM+1))+DD-694038
150 T=T/36525
160 LA=350.737486+1236*T*360
170 LA=LA+307*T+6*T/60
180 LA=LA+51.18*T/3600-5.17*T*T/3600
190 LA=LA-INT(LA/360)*360
200 LA=INT(LA+.5)
210 RETURN
220 REM SORTIE*****
230 PRINT
240 PRINT "PHASE LUNAIRE=";LA;"DEGRES"
250 PRINT "APPUYEZ SUR N'IMPORTE QUELLETOUCHE"
260 CALL KEY(5,K,S)
270 IF S=0 THEN 260
280 RETURN
290 REM PROGRAMME PRINCIPAL****
300 GOSUB 30
310 GOSUB 130
320 GOSUB 220
330 GOTO 300
340 END
```


Achevé d'imprimer
sur les presses de sofiacparis
8, rue de Furstenberg, 75006 Paris
Dépôt légal n° 6072 — mars 1984
Imprimé en France

Les grands classiques du jeu pour votre

TI 99/4A

Exercez-vous assez votre cerveau ? Votre esprit est-il aussi délié et vos réactions aussi rapides qu'ils devraient l'être ?

Tonifiez votre esprit grâce à ce recueil de quelques-uns des plus célèbres jeux jamais produits et que l'on ne rencontre traditionnellement que sur les gros ordinateurs.

- Battez-vous contre les **Monstres intergalactiques**,
- participez à l'excitante course de **Formule 1**,
- défendez-vous au cours d'une **Bataille navale**,
- et partez pour une palpitante **Chasse au requin**.

Si vous survivez à tous ces événements, vous pourrez

- connaître la **Fièvre à Las Vegas**,
- sonder les mystères de la **Boîte noire**,
- vous livrer à des jeux de réflexion comme **Mini Mancala** (inspiré d'un très ancien jeu arabe),
- ou à des jeux de logiques comme **ABBA**,
- et peut-être devenir champion de **Chiffres sans les lettres**.

Vous trouverez de très nombreux autres jeux pour vous-même, votre famille et tout votre entourage. Alors, à vos claviers ! Vous découvrirez des possibilités de votre machine que vous n'auriez jamais osé imaginer !



ADDISON-WESLEY EUROPE



InterEditions

87, AVENUE DU MAINE 75014 PARIS

